Κεφάλαιο 10

Γραφικές Παραστάσεις στο Επίπεδο

Στο κεφάλαιο αυτό θα παρουσιάσουμε αναλυτικά τις γραφικές παραστάσεις συναρτήσεων στο επίπεδο, οι οποίες παρουσιάζονται σε διάφορες μορφές, και συστήματα συντεταγμένων. Ακόμα, θα δούμε πώς δημιουργούνται τα βασικά σχήματα στο επίπεδο και πώς μπορούμε να δημιουργήσουμε σχήματα με κίνηση στις δύο διαστάσεις.

10.1 Γραφική Παράσταση Συνάρτησης

Θα ξεκινήσουμε με την ήδη γνωστή από προηγούμενα κεφάλαια εντολή `plot`, παρουσιάζοντας αρκετές λεπτομέρειες για τις παραμέτρους που αυτή δέχεται.

Περιγραφή:	***Δημιουργεί τη Γραφική Παράσταση μιας έκφρασης-συνάρτησης.***	
Γενική σύνταξη:	***plot(expr,h,v,opt)***	
Παράμετροι:	expr:	Η έκφραση ή η συνάρτηση της οποίας θα δημιουργήσουμε τη γραφική παράσταση.
	h:	Οριζόντιο διάστημα.
	v:	Κάθετο διάστημα.

Αν θέλουμε τη γραφική παράσταση περισσότερων συναρτήσεων στο ίδιο σύστημα αξόνων, μπορούμε να γράψουμε την εντολή στη μορφή `plot([f,g],h,v, opt);`

Παράδειγμα: 10-1

Να γίνει η γραφική παράσταση της συνάρτηση x^2 στο διάστημα [-2,2].

Λύση:

Ορίζουμε τη συνάρτηση f.

```
> f:=x->x^2;
```

$$f := x \to x^2$$

Ζητάμε αρχικά τη γραφική της παράσταση χωρίς να ορίζουμε διάστημα για το x και στη συνέχεια ορίζοντας διάστημα για το x το [-2,2].

```
> plot(f(x),x);
> plot(f(x),x=-2..2);
```





Αν θέλουμε να φανεί και ο αρνητικός ημιάξονας y, τότε μπορούμε να δώσουμε τιμές για το y.

```
> plot(f(x),x=-2..2,y=-4..4);
```



□

Η εντολή `plot` δέχεται επιπλέον και άλλες παραμέτρους, που μπορούν να μας βοηθήσουν να δημιουργήσουμε όμορφες γραφικές παραστάσεις.

Παράμετρος		**Επιλογές**
opt:	color=	Ορίζει το χρώμα της γραφικής παράστασης. *Δυνατές τιμές :aquamarine, black, blue, navy, coral, cyan brown, gold, green, gray, grey, khaki magenta, maroon, orange, pink, plum, red, sienna, white, yellow*. προεπιλογή είναι: black
	axes =	Ορίζει τον τύπο των αξόνων που θα εμφανίζονται. *Δυνατές τιμές: frame, boxed,normal,none*
	style=	Ορίζει αν η γραφική παράσταση θα σχηματισθεί από σημεία, συνεχόμενη γραμμή ή άλλο σύμβολο. *Δυνατές τιμές: line, point, patch, patchnogrid.* προεπιλογή είναι η line.
	labels= [x, y]	Ορίζει το κείμενο που θα εμφανίζεται στους άξονες.
	title=	Ο τίτλος της γραφικής παράστασης.
	legend=s	Όταν έχουμε περισσότερες από μία καμπύλες, τότε δημιουργεί υποσημείωση για κάθε μία από αυτές.
	symbolsize=	Το μέγεθος των συμβόλων που χρησιμοποιούνται. προεπιλογή είναι:n=10

Πίνακας 16. *Παράμετροι της εντολής plot.*

Παράδειγμα: 10-2

Να γίνει η γραφική παράσταση της συνάρτησης $\sin(x)$ στο διάστημα [-2π,2π].

Λύση:

Στο παράδειγμα αυτό θα σχηματίσουμε τη γραφική παράσταση χρησιμοποιώντας σημεία (`style=point`) χρώματος μπλε (`color=navy`) και οι άξονες θα είναι στη μορφή κουτιού (`axes=boxed`), ενώ θα δώσουμε και τίτλο στη γραφική παράσταση.

```
> plot(sin(x),x=-2*Pi..2*Pi,color=navy, title="Ημίτονο",
style=point,axes=boxed);
```



Μπορούμε επίσης σχηματίσουμε την γραφική παράσταση χρησιμοποιώντας σημεία (`style=point`) χρώματος κόκκινου (`color=red`), αυτή τη φορά μεγαλύτερου μεγέθους, (`symbolsize=30`) και άξονες στη μορφή frame (`axes=frame`),

```
>plot(sin(x),x=-2*Pi..2*Pi,title="Ημίτονο",axes= frame,
style=point,symbolsize=30,labels= [time,height]);
```



□

Στον παρακάτω πίνακα ακολουθούν κάποιες επιπλέον παράμετροι της εντολής `plot`.

Παράμετρος		Επιλογές
opt	coords=	Ορίζει το σύστημα των συντεταγμένων που θα χρησιμοποιηθεί. Δυνατές τιμές: *bipolar, cardioid, cassinian, cartesian, elliptic, hyperbolic, invcassinian, invelliptic, logarithmic, logcosh, maxwell, parabolic, polar, rose, tangent.* προεπιλογή είναι: Cartesian.
	filled=	Ορίζει εάν η περιοχή ανάμεσα στην καμπύλη της συνάρτησης και των άξονα x θα είναι «γεμάτη» με χρώμα ή όχι. Δυνατές τιμές: *true, false*
	tickmarks=[m, n]	Ορίζει τον αριθμό τον σημείων που θα εμφανίζονται στους άξονες.
	xtickmarks=	Ο αριθμός των σημείων που θα εμφανίζονται στον άξονα x.
	ytickmarks=	Ο αριθμός των σημείων που θα εμφανίζονται στον άξονα y.
	discont=	Ορίζει αν θα εμφανίζονται οι ασυνέχειες. Δυνατές τιμές: *true, false* προεπιλογή είναι: *false*
	numpoints=	Ορίζει τον ελάχιστο αριθμό σημείων που θα δημιουργούν τη γραφική παράσταση. προεπιλογή είναι:50
	resolution=n	Ορίζει την οριζόντια ανάλυση σε pixels. προεπιλογή είναι:n=200
	scaling	Ορίζει την κλίμακα. Δυνατές τιμές: constrained, unconstrained. προεπιλογή είναι:unconstrained.

opt	thickness=	Ορίζει το πάχος της γραμμής. προεπιλογή είναι: 0.
	adaptive=	Ορίζει το αν υποδιαιρεθούν τα διαστήματα χάραξης της γραφικής παράστασης ώστε να έχουμε καλύτερο αποτέλεσμα. Δυνατές τιμές:true, false προεπιλογή είναι true.

Πίνακας 15. *Παράμετροι της εντολή plot για την ευκρίνεια.*

Παράδειγμα: 10-3

Να γίνει η γραφική παράσταση της συνάρτηση $\sin(x)$ στο διάστημα [0,2π].

Λύση:

Με την παράμετρο `coords` ορίζουμε το σύστημα συντεταγμένων. Ας δούμε τη γραφική παράσταση του $\sin(x)$ σε πολικές συντεταγμένες. Για τη γραφική παράσταση σε πολικές συντεταγμένες υπάρχει και η εντολή `polarplot` που θα συναντήσουμε στη συνέχεια.

```
> plot(sin(x),x=0..2*Pi, coords=polar);
```



Θα χρησιμοποιήσουμε άλλες δυο παραμέτρους του πίνακα 15, την παράμερο `tickmarks`, για να εμφανίζονται οι τιμές 10 σημείων στον οριζόντιο και στον κάθετο άξονα, και την παράμετρο `thickness`, για να αυξήσουμε το πάχος της γραμμής.

```
>plot(sin(x),x=0..2*Pi,coords=polar,tickmarks=
[10,10],thickness=5);
```



Παράδειγμα: 10-4

Να γίνει η γραφική παράσταση της συνάρτησης $f(x)=\begin{cases}-1, \text{αν } x\le 1\\ x \quad \text{αν } x>1\end{cases}$.

Λύση:

Στο παράδειγμα αυτό έχουμε μια ασυνεχή πολύκλαδη συνάρτηση.

Ορίζουμε την f.

```
> f:= piecewise( x<=1, -1, x>1,x );
```

$$f := \begin{cases} -1 & x \le 1 \\ x & 1 < x \end{cases}$$

Ζητάμε τη γραφική παράστασή της στο διάσημα [-2,2].

```
> plot(f(x),x=-2..2, thickness=3);
```



Βλέπουμε ότι εκεί όπου έχουμε ασυνέχεια το Maple ενώνει με μια κάθετη γραμμή. Για να μη γίνεται αυτό, προσθέτουμε την παράμετρο `discont=true`.

```
> plot(f(x),x=-2..2,discont=true);
```



Ας δούμε μια ακόμα γραφική παράστασης δίκλαδης συνάρτησης, της γνωστής Heaveside συνάρτησης.

```
> plot(Heaviside(x),x,thickness=3,title="Heaviside
function",discont=true);
```



Σε αυτό το σημείο θα χρησιμοποιήσουμε την εντολή `convert`, για να μετατρέψουμε την έτοιμη εντολή υλοποίησης της συνάρτησης `Heaviside` σε πολύκλαδη.

```
> convert(Heaviside(x),piecewise,x);
```

$$\begin{cases} 0 & x < 0 \\ undefined & x = 0 \\ 1 & 0 < x \end{cases}$$

□

Παράδειγμα: 10-5

Να γίνει η γραφική παράσταση της συνάρτησης $f(x) = \tan x$.

Λύση:

Ορίζουμε τη συνάρτηση.

```
> f:=x->tan(x);
```

$$f := x \to \tan(x)$$

Αν ζητήσουμε τη γραφική της παράσταση, το αποτέλεσμα που θα πάρουμε δε θα είναι καθόλου ικανοποιητικό. Βλέπουμε ότι:

```
> plot(f(x),x=-2*Pi..2*Pi);
```



Αν δώσουμε διάστημα και στις τιμές y, τότε:

```
> plot(f(x),x=-2*Pi..2*Pi,y=-4..4);
```



Ένα καλύτερο αποτέλεσμα θα έχουμε, αν θέσουμε την παράμετρο `discont=true`. Τότε δεν εμφανίζονται οι ασύμπτωτες (κάθετες γραμμές).

```
> plot(f(x),x=-2*Pi..2*Pi,y=-6..6,discont=true,
xtickmarks=[-6.28="2p",-4.71="-3p/2",-3.14="-p",-1.57="-
p/2",1.57="p/2",3.14="p",4.71="-3p/2",6.28="2p"],
axesfont=[SYMBOL,12]);
```



Παράδειγμα: 10-6

Να γίνει η γραφική παράσταση της συνάρτησης $f(x) = x^2 - 2$ και να γραμμοσκιαστεί η περιοχή που περικλείεται από τη συνάρτηση και τον άξονα $x'x$.

Λύση:

Με την παράμετρο `filled` μπορούμε να ορίσουμε, αν θέλουμε, γέμισμα στην περιοχή που περικλείεται από τη συνάρτηση και τον άξονα των x.

```
> plot(x^2-2,x=-2..2,y=-4..4,filled=true, color=[blue]);
```



<table>
<tr><th colspan="2">Παράμετρος</th><th>Επιλογές</th></tr>
<tr><td rowspan="5">opt</td><td>font=
[family, style, size]</td><td>Ορίζει τον τύπο της γραμματοσειράς για κείμενο στην εντολή plot.
family=TIMES,COURIER,HELVETICA,SYMBOL.
style= ROMAN, BOLD, ITALIC, BOLDITALIC.
size= το μέγεθος της γραμματοσειράς.</td></tr>
<tr><td>axesfont=</td><td>Ορίζει τον τύπο της γραμματοσειράς για τους άξονες</td></tr>
<tr><td>titlefont=</td><td>Ορίζει τον τύπο της γραμματοσειράς για την παράμετρο title.
Δυνατές τιμές: εκείνες της παραμέτρου font.</td></tr>
<tr><td>labelfont=</td><td>Ορίζει τον τύπο της γραμματοσειράς για την παράμετρο labels.</td></tr>
<tr><td>labeldirections=
[x, y]</td><td>Ορίζει την κατεύθυνση του κειμένου στους άξονες, το οποίο ορίσθηκε από την παράμετρο labels.Δυνατές τιμές: horizontal, vertical. Προεπιλογή: horizontal.</td></tr>
</table>

Πίνακας 16. *Παράμετροι της εντολής plot για κείμενο.*

Παράδειγμα: 10-7

Να γίνει η γραφική παράσταση της συνάρτησης $f(x) = x^2 \sin^2(x)$ στο διάστημα [-5π,5π].

Λύση:

Εδώ θα ορίσουμε πρώτα τη συνάρτηση f.

```
> f := x -> x^2 * sin(x)^2;
```

$$f := x \to x^2 \sin(x)^2$$

```
>plot( f(x), x = -5*Pi .. 5*Pi,font=[COURIER, BOLDOBLIQUE
,10],titlefont=[HELVETICA,    BOLDOBLIQUE,   16],title="H
συνάρτηση x^2*sin(x)");
```



10.2 Οδηγός Γραφικής Παράστασης

Ένα ακόμα χαρακτηριστικό της έκδοσης 11 του Maple είναι η δυνατότητα που δίνει να σχηματίσουμε γραφικές παραστάσεις χρησιμοποιώντας τις ευκολίες ενός «οδηγού».

Αν στο φύλλο εργασίας ενεργοποιήσουμε αρχικά το πακέτο `plots` και χρησιμοποιήσουμε την εντολή `interactive`, τότε μπορούμε να ενεργοποιήσουμε τον οδηγό γραφικών παραστάσεων.

```
> with(plots):
> interactive(sin(x));
```





Μέσα από αυτόν τον οδηγό μπορούμε να ρυθμίσουμε τις διάφορες παραμέτρους της γραφικής παράστασης που ζητάμε. Χρησιμοποιώντας στη συνέχεια την επιλογή `plot` έχουμε:



□

10.3 Γραφική Παράσταση Πολλών Συναρτήσεων

Στο παρακάτω παράδειγμα χρησιμοποιούμε την εντολή `plot` για να δημιουργήσουμε στο ίδιο σύστημα αξόνων τη γραφική παράσταση δύο συναρτήσεων.

Παράδειγμα: 10-8

Να γίνουν στο ίδιο σύστημα αξόνων οι γραφικές παραστάσεις των συναρτήσεων $\cos(x)$ και x^2.

Λύση:

```
> plot([sin(x),cos(x)],x=-2*Pi..2*Pi, colour=[blue, red], title= "Τριγωνομετρικές Συναρτήσεις Sine - Cosine" , titlefont=[TIMES,ITALIC,18], legend=["Sine", "Cosine"]);
```



□

Παράδειγμα: 10-9

Να γίνουν στο ίδιο σύστημα αξόνων οι γραφικές παραστάσεις των συναρτήσεων $4-x^2$ και x^2 και να γραμμοσκιαστεί η μεταξύ τους περιοχή.

Λύση:

Για τη γραμμοσκίαση της περιοχής μεταξύ των δύο συναρτήσεων χρησιμοποιούμε την εντολή `plot` με την παράμετρο `filled=true`.

```
>  plot([x^2,4-x^2],x=-2..2,filled=true,color=[COLOR(RGB,
.9,.9,.99), blue],scaling=constrained);
```



□

10.4 Σχεδιάζοντας Σημεία

Στο κεφάλαιο 3 είδαμε πώς μπορούμε να παραστήσουμε σημεία με την εντολή `listplot` του πακέτου plots. Τώρα θα δούμε πώς μπορούμε να έχουμε το ίδιο αποτέλεσμα με τη χρήση της εντολή `plot`.

Περιγραφή:	***Σχεδιάζει Σημεία στο επίπεδο.***	
Γενική σύνταξη:	***plot([[x1,y1],[x2,y2],…], opt)***	
Παράμετροι:	[[x1,y1], [x2,y2],…]	Οι συντεταγμένες σημείων δίνονται μέσα σε τετράγωνες αγκύλες και τα σημεία χωρίζονται με κόμμα.
	symbol=	Το σύμβολο που χρησιμοποιείται για την αναπαράσταση του σημείου: BOX, CROSS, CIRCLE, POINT,DIAMOND.

		προεπιλογή είναι: POINT
	symbolsize=	Το μέγεθος των συμβόλων που χρησιμοποιούνται. προεπιλογή είναι:n=10
	opt	Ό,τι και στην εντολή `plot`.

Παράδειγμα: 10-10

Να παρασταθούν στο επίπεδο τα σημεία $(0.5,1.2)$, $(1,1.5)$, $(1.2,2)$, $(-0.3,-0.5)$, $(-0.3,0.5)$.

Λύση:

Χρησιμοποιώντας την εντολή `plot` σχεδιάζουμε τα σημεία, χρώματος μπλε, μεγέθους 20 και σχήματος διαμαντιού.

```
> plot([[0.5,1.2],[1,1.5],[1.2,2],[-0.3,-0.5],[-.3,0.5]],
style=point,color=blue,symbol=diamond,symbolsize=20);
```



Αν στην παραπάνω εντολή, αντί για `style=point`, χρησιμοποιήσουμε `style=line`, το αποτέλεσμα θα είναι μια τεθλασμένη γραμμή, η οποία διέρχεται από τα σημεία αυτά ξεκινώντας από αυτό που παραθέτουμε πρώτο και τελειώνοντας στο τελευταίο.

```
> plot([[0.5,1.2],[1,1.5],[1.2,2],[-0.3,-0.5],[-.3,0.5]],
style=line,color=blue);
```



10.5 Σημεία και Γραφικές Παραστάσεις.

Αν θέλουμε να συνδυάσουμε επιμέρους γραφικές παραστάσεις, πρέπει να χρησιμοποιήσουμε την εντολή `display` του πακέτου εντολών `plots`.

Περιγραφή:	***Εμφανίζει στο ίδιο σύστημα αξόνων δύο ή περισσότερες γραφικές παραστάσεις.***
Γενική σύνταξη:	***display(εικόνα1, εικόνα2,...)***
Πακέτο:	***plots***

Η εντολή `display` παίρνει τις παραμέτρους της εντολής `plots`.

Παράδειγμα: 10-11

Να παρασταθούν στο ίδιο σύστημα αξόνων η γραφική παράσταση της παραβολής x^2-9, της ευθείας $-3x+5$, καθώς και των σημείων τομής τους που είναι [-1,8], [4,-7].

Λύση:

```
> with(plots):
```

Ορίζουμε ως pict1 τη γραφική παράσταση της παραβολής και της ευθείας.

```
> pict1:=plot([-3*x+5,9-x^2],x=-3..5,color=[green, red]):
```

Ορίζουμε ως pict2 τα σημεία [-1,8], [4,-7], τα οποία ζητάμε να είναι χρώματος μπλε και να παριστάνονται με μικρούς κύκλους.

```
> pict2:=plot([[-1,8],[4,-7]],style=point,color= blue, symbol=circle):
```

Ζητάμε να εμφανιστούν στο ίδιο σύστημα αξόνων οι δύο εικόνες pict1 και pict2.

```
> display(pict1,pict2);
```



Αν θέλουμε σε μια γραφική παράσταση να προσθέσουμε κείμενο σε οποιαδήποτε θέση της, μπορούμε να χρησιμοποιήσουμε την εντολή `textplot`.

Περιγραφή:	***Εμφανίζει κείμενο σε προκαθορισμένη θέση.***
Γενική σύνταξη:	***textplot([x,y,"κείμενο"])***
Πακέτο:	***plots***

Ορίζουμε ως txt1 το κείμενο που θέλουμε να εμφανιστεί στη θέση [-1,8].

```
> txt1:=textplot([-1,8,"1o σημείο τομής"]):
```

Όμοια, ορίζουμε το txt2.

```
> txt2:=textplot([4,-7,"2o Σημείο τομής"]):
```

Ζητάμε να εμφανιστούν στο ίδιο σύστημα αξόνων οι δύο εικόνες pict1 και pict2 και τα txt1 και txt2.

```
> display(pict1,pict2,txt1,txt2);
```



□

10.6 Γραφική Παράσταση Πεπλεγμένης Συνάρτησης

Για τη γραφική παράσταση συνάρτησης που βρίσκεται σε πεπλεγμένη μορφή, χρησιμοποιούμε την εντολή `implicitplot` του πακέτου εντολών `plots`.

Περιγραφή:	***Δημιουργεί τη γραφική παράσταση μιας έκφρασης-συνάρτησης που βρίσκεται σε πεπλεγμενη μορφή.***	
Γενική σύνταξη:	**`implicitplot`** ***(expr, x=a..b, y=c..d, options)***	
Παράμετροι:	expr:	Η έκφραση-συνάρτηση.
	x=a..b	Το διάστημα για τη μεταβλητή x.
	y=c..d	Το διάστημα για τη μεταβλητή y.
	options:	Ότι και στην εντολή plot.
Πακέτο:	***plots***	

Παράδειγμα: 10-12

Θα δώσουμε τις γραφικές παραστάσεις δύο συναρτήσεων που βρίσκονται σε πλεγμένη μορφή $x^2+y^2=1$ και $x^2+y^4=y^2-x$.

Λύση:

Ενεργοποιούμε το πακέτο plots.

```
> with( plots ):
```

Χρησιμοποιούμε την εντολή `implicitplot` στο διάστημα [-1..1] για το x και το y.

```
> implicitplot(x^2+y^2=1,x=-1..1,y=-1.. 1,scaling=
constrained );
```



Χρησιμοποιούμε την εντολή `implicitplot` στο διάστημα [-1.3,0.3] για το x και το [-1.2,1.2] για το y.

```
> implicitplot(x^2+y^4=y^2-x,x=-1.3..0.3,y=-1.2..
1.2,scaling=constrained );
```



□

10.7 Γραφική Παράσταση Παραμετρικών Εξισώσεων

Για να σχηματίσουμε τη γραφική παράσταση παραμετρικών εξισώσεων της μορφής $\begin{cases} x = f(t) \\ y = g(t) \end{cases}, t \in [a,b]$, θα χρησιμοποιήσουμε την εντολή `plot` ως εξής:

Περιγραφή:	*Γραφική Παράσταση παραμετρική συνάρτηση.*
Γενική σύνταξη:	*plot([f(t),g(t) ,t=a..b], παράμετροι)*
Παράμετροι:	Ό,τι και στην εντολή plot

Παράδειγμα: 10-13

Να γίνει η γραφική παράσταση της καμπύλης που ορίζεται από $\begin{cases} x = t^2 - t \\ y = 2t - t^3 \end{cases}, t \in [-2,2]$.

Λύση:

Η εντολή Plot είναι:

```
> plot([t^2-t,2*t-t^3,t=-2..2] ,thickness=3);
```



□

Παράδειγμα: 10-14

Να γίνει η γραφική παράσταση της καμπύλης που ορίζεται

από $\begin{cases} x = \sin 3t \\ y = \sin 4t \end{cases}, t \in [0, 2\pi]$.

Λύση:

Η εντολή `plot` για τις παραμετρικές εξισώσεις είναι:

```
> plot([sin(3*t),sin(4*t),t=0..2*Pi],title= "x=sin3t,
y=sin4t" ,thickness=3);
```



Αν θέλουμε να φαίνεται μεγαλύτερο διάστημα στους άξονες, τότε μπορούμε να προσθέσουμε το εύρος μέσα στο οποίο μεταβάλλεται το x και το y.

```
> plot([sin(3*t),sin(4*t),t=0..2*Pi],x=-2..2,y=-2..2);
```

10.8 Γραφική Παράσταση Συνάρτησης σε Πολικές Συντεταγμένες

Όπως είδαμε, η εντολή plot έχει ως προεπιλεγμένο σύστημα συντεταγμένων τις καρτεσιανές συντεταγμένες. Αν εμείς θελήσουμε να χρησιμοποιήσουμε πολικές συντεταγμένες, πρέπει στην εντολή `plot` να θέσουμε την παράμετρο `coords=polar` ή να χρησιμοποιήσουμε την εντολή `polarplot` του πακέτου plots.

Περιγραφή:	***Δημιουργεί τη γραφική παράσταση μιας έκφρασης-συνάρτησης σε πολικές συντεταγμένες.***	
Γενική σύνταξη:	***polarplot(expr, angle=range)***	
Παράμετροι:	expr	Η εξίσωση εκφρασμένη ως προς τη γωνία.
	angle	Το εύρος διακύμανσης της γωνίας.
Πακέτο:	***plots***	

Παράδειγμα: 10-15

Να σχεδιαστεί ο κύκλος με εξίσωση $r = 1$.

Λύση:

Ενεργοποιούμε το πακέτο `plots`.

```
> withs(plots):
> polarplot(1,theta=0..2*Pi );
```



Παράδειγμα: 10-16

Στο παράδειγμα αυτό θα δούμε κάποιες γραφικές παραστάσεις συναρτήσεων που δίνονται σε πολικές συντεταγμένες.

Λύση:

Ενεργοποιούμε το πακέτο `plots`.

```
> with(plots):
```

- Η γραφική παράσταση του καρδιοειδούς $r = 1 + \cos\theta$ είναι:

```
> polarplot(1+cos(theta),theta=-Pi..Pi,scaling=
constrained, thickness=3);
```



- Η γραφική παράσταση $r = \sin 3\theta$ είναι:

```
> polarplot(sin(3*theta),theta=-Pi..Pi, scaling=
constrained, thickness=3 );
```



- Η γραφική παράσταση της σπείρας $r = \theta$ είναι:

```
> polarplot(theta,theta=0..8*Pi,thickness=3);
```



10.9 Γραφικές Παραστάσεις Βασικών Σχημάτων στο Επίπεδο

Στην παράγραφο αυτή θα παρουσιάσουμε τις εντολές εκείνες που μας επιτρέπουν να κατασκευάσουμε τα βασικότερα σχήματα στο επίπεδο. Για τη χρήση αυτών των εντολών είναι απαραίτητο να ενεργοποιηθούν τα πακέτα `plots` και `plottools`. Παρακάτω φαίνεται ποιες εντολές ενεργοποιεί καθένα από τα πακέτα αυτά.

```
> with(plots);
```

[animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d,
conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d,
cylinderplot, densityplot, display, display3d, fieldplot, fieldplot3d, gradplot,
gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, listcontplot,
listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot,
odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot,
polygonplot3d, polyhedra_supported, polyhedraplot, replot, rootlocus, semilogplot,
setoptions, setoptions3d, spacecurve, sparsematrixplot, sphereplot, surfdata, textplot,
textplot3d, tubeplot]

```
> with(plottools);
```

[arc, arrow, circle, cone, cuboid, curve, cutin, cutout, cylinder, disk, dodecahedron, ellipse, ellipticArc, hemisphere, hexahedron, homothety, hyperbola, icosahedron, line, octahedron, pieslice, point, polygon, project, rectangle, reflect, rotate, scale, semitorus, sphere, stellate, tetrahedron, torus, transform, translate, vrml]

Παρατίθενται ακολούθως οι εντολές με τις οποίες μπορούν να σχηματιστούν τα βασικότερα σχήματα. Για να εμφανισθούν στην οθόνη τα σχήματα αυτά, από τη στιγμή που θα τα ορίσουμε, απαιτείται η χρήση της εντολής `display`.

ΣΧΗΜΑ	ΕΝΤΟΛΗ
ΚΥΚΛΟΣ	***circle(c, r, options);***
ΕΛΛΕΙΨΗ	***ellipse(c, a, b, options);***
ΥΠΕΡΒΟΛΗ	***hyperbola(c, a, b, r, options);***
ΤΕΤΡΑΓΩΝΟ	***Rectangle([x1, y1], [x2, y2], options)***
ΔΙΣΚΟΣ	***disk([x, y], r, options)***
ΤΟΞΟ	***arc(c, r, a..b, options)***
ΒΕΛΟΣ	***arrow(base, dir, wb, wh, hh, sh)***
ΣΗΜΕΙΑ	***point(a, options)***
ΓΡΑΜΜΗ	***line(a, b, options)***

Πίνακας 17. *Βασικά Σχήματα στο Επίπεδο.*

Παράδειγμα: 10-17

Να σχεδιαστεί κύκλος με κέντρο στο σημείο (0,0) και ακτίνα το 1.5.

Λύση:

```
> with(plottools):with(plots):
> cir:=circle([0,0],1.5,color=green):
display(cir,scaling=CONSTRAINED,view=[-2..2,-2..2],
title="Κύκλος με κέντρο το (0,0)");
```



□

Παράδειγμα: 10-18

Να σχεδιαστεί έλλειψη με κέντρο το σημείο (0,0), μεγάλο άξονα το 2 και μικρό άξονα το 1.

Λύση:

```
> with(plottools):with(plots):
> elli:=ellipse([0,0],2,1,color=green):
> display(elli,scaling=CONSTRAINED,view=[-2..2,-2..2],
title="Έλλειψη με κέντρο το (0,0)");
```



Αν θέλουμε η έλλειψη να είναι γραμμοσκιασμένη, τότε χρησιμοποιούμε την παράμετρο `filled=true`.

```
> with(plottools):with(plots):
> elli:=ellipse([0,0],2,1,color=blue,filled=true):
```

```
> display(elli,scaling=CONSTRAINED,view=[-2..2,-2..2],
title="Ελλειψη με κέντρο το (0,0)");
```



□

Παράδειγμα: 10-19

Να σχεδιαστεί η υπερβολή με κέντρο το (0,0) και μεγάλο άξονα 1,2.

Λύση:

```
> with(plottools):with(plots):
```

Ορίζουμε ως hyp την υπερβολή με κέντρο το (0,0) και μεγάλο άξονα το 1,2.

```
> hyp:=hyperbola([0,0], 1,2, -2..2):
```

Ζητάμε να εμφανιστεί η υπερβολή δίνοντας επιπλέον και έναν τίτλο στη γραφική αυτή παράσταση.

```
> display(hyp,scaling=CONSTRAINED,view=[-3..3,-3..3],
title="Υπερβολή με κέντρο το (0,0)");
```



□

Παράδειγμα: 10-20

Να σχεδιαστεί τετράγωνο με κάτω αριστερή γωνία στο σημείο (0,1) και πάνω δεξιά το (2,4).

Λύση:

```
> with(plottools):with(plots):
> display(rectangle([0,1], [2,4], color=blue));
```



□

Παράδειγμα: 10-21

Να σχεδιαστεί κυκλικός δίσκος με κέντρο το σημείο (1,1) του επιπέδου, ακτίνα 5, και χρώματος κόκκινου.

Λύση:

```
> with(plottools):with(plots):
> display(disk([1,1],5,color=red));
```



□

Παράδειγμα: 10-22

Να σχεδιαστεί ένα τόξο με κέντρο κύκλου το σημείο (0,0), ακτίνα του κύκλου ίση με 1, αρχική γωνία $\frac{\pi}{2}$ και τελική 2π .

Λύση:

```
> with(plottools):with(plots):
> display(arc([0,0], 1, Pi/2..2*Pi));
```



□

Παράδειγμα: 10-23

Να σχεδιαστεί βέλος χρώματος πράσινου με αρχή το σημείο (0,0) και πέρας το σημείο (10,10), πάχος γραμμής 0.2 και χαρακτηριστικά πέρατος 0.4 και 0.2 .

Λύση:

```
> with(plottools):with(plots):
>display(arrow([0,0],[10,10],0.2,0.4,0.1,color=green));
```

□

Παράδειγμα: 10-24

Να σχεδιαστεί γραμμή με αρχή το σημείο (0,1) και πέρας το σημείο (2,3), χρώματος κόκκινου.

Λύση:

```
> with(plots):with(plottools):
> display(line([0,1], [2,3], color=red));
```



□

10.10 Κίνηση

Η κίνηση είναι μια ακολουθία από εικόνες οι οποίες εμφανίζονται διαδοχικά στην οθόνη του υπολογιστή. Κινούμενες εικόνες στο Maple μπορούμε να δημιουργήσουμε με την εντολή `animate` του πακέτου `plots`.

Περιγραφή:	***Δημιουργεί μια κινούμενη εικόνα***		
Γενική σύνταξη:	***animate(plotcommand, plotargs, t=a..b, options)***		
Παράμετροι:	plotcommand	Η εντολή για την οποία θα δημιουργήσουμε κίνηση.	
	plotargs	Οι παράμετροι της παραπάνω εντολής.	
	t= a..b	Το εύρος της κίνησης.	
	options	frames=	Αριθμός των εικόνων, (προεπιλογή είναι:25)
		background=	Η εικόνα που βρίσκεται στο φόντο.
Πακέτο:	***plots***		

Παράδειγμα: 10-25

Στο παράδειγμα αυτό θα δημιουργήσουμε μια κινούμενη εικόνα που θα αποτελείται από τη γραφική παράσταση του συνημιτόνου την οποία θα διατρέχει ένα σημείο. Επίσης θα δούμε την ακολουθία εικόνων από την οποία δημιουργείται η κινούμενη εικόνα .

Λύση:

Ενεργοποιούμε το πακέτο plots.

```
> with(plots):
```

Χρησιμοποιούμε την εντολή `animate` με `plotcommand` την εντολή `plot`

```
>animate(plot,[[[x,cos(x)]],style=point,symbolsize=30,col
or=black], x=0..4*Pi, background=plot(cos(x),x=0..4*Pi),
frames=8);
```



Για να δούμε τα 8 frames, από τα οποία αποτελείται η κινούμενη εικόνα χρησιμοποιούμε την εντολή `display`.

```
> display(%);
```



Παράδειγμα: 10-25

Στο παράδειγμα αυτό θα δημιουργήσουμε έναν κύκλο με κέντρο το (0,0) στον οποίο θα μεταβάλλεται η ακτίνα του από -2..2.

Λύση:

Ενεργοποιούμε το πακέτο `plots`.

```
> with(plots):
```

Χρησιμοποιούμε την εντολή `animate` με `plotcommand` την εντολή `implicitplot`.

```
> animate(implicitplot,[x^2+y^2=a^2
,x=-2..2,y=-2..2],a=-2..2);
```



Παράδειγμα: 10-26

Στο παράδειγμα αυτό θα δημιουργήσουμε μια κινούμενη εικόνα στην οποία θα εμφανίζεται η γραφική παράσταση του συνημιτόνου και της παραγώγου του, καθώς και η εφαπτομένη σε κάθε σημείο από 0 έως 2π.

Λύση:

Ενεργοποιούμε το πακέτο plots.

```
> restart: with(plots):
```

Ορίζουμε τη συνάρτηση $f(x) = \cos(x)$.

```
>f := x -> cos(x):
```

Ορίζουμε την συνάρτηση της πρώτης παραγώγου της $f(x)$.

```
>df:= x -> D(f)(x):
```

Ως p1 ορίζουμε να είναι η κινούμενη εικόνα της $f(x)$ καθώς την διατρέχει ένα σημείο της

```
>p1:= animate(plot,[[[x,f(x)]], style= point,
symbolsize= 15, color=black],x=0..2*Pi,view=-2..2,
background=plot(f(x),x=0..2*Pi,color=red),frames=80)
```

Ως p2 ορίζουμε να είναι η κινούμενη εικόνα της εφαπτόμενης ευθείας σε κάθε σημείο της $f(x)$

```
>p2  :=  animate(  df(x0)*(x-x0)+f(x0),  x=0..2*Pi,  x0=
0..2*Pi ,color=black,frames=80):
```

Ως p3 ορίζουμε να είναι η κινούμενη εικόνα της παραγώγου της $f(x)$

```
>p3 := animate( [x*x0/(2*Pi),df(x*x0/(2*Pi)),
x=0..2*Pi],x0=0..2*Pi, color=blue,frames=80):
```

Εμφανίζουμε και τις τρεις κινούμενες εικόνες συγχρόνως.

```
>display(p1,p2,p3);
```



Κεφάλαιο 11

Λυμένες Ασκήσεις στις Συναρτήσεις μιας Μεταβλητής

Στο κεφάλαιο αυτό θα χρησιμοποιήσουμε τις εντολές που είδαμε μέχρι τώρα, για να λύσουμε κάποιες χαρακτηριστικές ασκήσεις στις συναρτήσεις μιας μεταβλητής.

Άσκηση: 11-1

Να βρεθούν τα κοινά σημεία του κύκλου $x^2+y^2=1$ και της ευθείας $x+2y=1$ και να παρασταθούν στο ίδιο σύστημα αξόνων.

Λύση:

Ορίζουμε ως eq1 και eq2 τις δύο εξισώσεις.

```
> eq1:=x^2+y^2=1;
```

$$eq1 := x^2+y^2=1$$

```
> eq2:=x+2*y=1;
```

$$eq2 := x + 2\,y = 1$$

Βρίσκουμε τα σημεία τομής τους λύνοντας το σύστημα των δύο εξισώσεων και τις λύσεις τις ονομάζουμε sol.

```
> sol:=[solve({eq1,eq2},{x,y})];
```

$$sol := \left[\ \{x = 1, y = 0\},\ \{x = \frac{-3}{5}, y = \frac{4}{5}\}\ \right]$$

Ορίζουμε τα σημεία ως pts.

```
> pts:=seq(eval([x,y],pt),pt=sol);
```

$$pts := [1, 0], \left[\frac{-3}{5}, \frac{4}{5}\right]$$

Φορτώνουμε το πακέτο plots.

```
> with(plots):
```

Ορίζουμε ως p1 τις γραφικές παραστάσεις των δύο εξισώσεων που είναι σε πεπλεγμένη μορφή.

```
> p1:=implicitplot({eq1,eq2},x =-2 .. 2,y =-2..2,
scaling=constrained ):
```

Ορίζουμε ως p2 τη γραφική αναπαράσταση των σημείων τομής τους.

```
> p2:=plot([pts],style=point,color=black, symbol=circle,
symbolsize=30 ):
```

Με την εντολή `display` παρουσιάζουμε στο ίδιο σύστημα αξόνων τις γραφικές παραστάσεις p1 και τα σημεία τομής τους p2.

```
> display([p1,p2]);
```



Άσκηση: 11-2

Να παρασταθεί γραφικά η σύγκλιση της συνάρτησης $f(x)=\sin(x)$ καθώς το $x\to 1$.

Λύση:

Χρησιμοποιώντας την εντολή `limit` βρίσκουμε:

```
> limit(sin(x),x=1);
```

$$\sin(1)$$

```
> evalf(%);
```

$$0.8414709848$$

Θα παραστήσουμε γραφικά τη συνάρτηση f και σημεία κοντά στο όριο.

```
> with(plots):
```

Ορίζουμε τη συνάρτηση f.

```
>f := x -> sin(x) ;
```

Ορίζουμε το σημείο x0.

```
>x0 := 1;
```

$$x0 := 1$$

Ορίζουμε ως p1 τη γραφική παράσταση της f.

```
> p1:=plot( f(x), x = 0..2,color=blue) :
```

$$f := x \to \sin(x)$$

Ορίζουμε ως p2 τα ευθύγραμμα τμήματα που δείχνουν τις συντεταγμένες του σημείου $\left(x_0, f(x_0)\right)$.

```
> p2:=plot( {[[x0,0],[x0,f(x0)]],[[0,f(x0)], [x0,f(x0)]]},
x = 0..2, linestyle=3,thickness = 2):
```

Ορίζουμε ως p3 σημεία που συγκλίνουν στο $\left(x_0, f(x_0)\right)$ από αριστερά.

```
> p3:= plot([[ x0 - 1/n, f(x0 - 1/n)] $n=1..30], x = 0..2,
style=point, symbol=circle, color = red):
```

Ορίζουμε ως p4 σημεία που συγκλίνουν στο $\left(x_0, f(x_0)\right)$ από δεξιά.

```
> p4:=plot([[ x0+1/n, f(x0 + 1/n)] $n=1..30],x = 0..2,
style=point, symbol=circle, color = green):
```

Με την εντολή `display` εμφανίζουμε όλες τις γραφικές παραστάσεις μαζί.

```
> display( p1,p2,p3,p4);
```



Άσκηση:11-3

Να εξετασθεί ως προς τη συνέχεια η συνάρτηση $f(x) = \dfrac{\sin(x)}{x}$.

Λύση:

Ορίζουμε τη συνάρτηση f.

```
> f:=x->sin(x)/x;
```

$$f := x \rightarrow \frac{\sin(x)}{x}$$

Θα αναζητήσουμε τα σημεία ασυνέχειας της συνάρτησης, αν υπάρχουν. Αυτό μπορεί να επιτευχθεί εύκολα με τη χρήση της εντολή `discont`, η οποία βρίσκει τα σημεία ασυνέχειας μιας πραγματικής συνάρτησης.

Γενική σύνταξη:	***discont(f(x),x)***

```
> discont(f(x),x);
```

$$\{0\}$$

Πράγματι, όπως ήταν αναμενόμενο η εντολή επιστρέφει την τιμή 0, αφού αυτό είναι το σημείο ασυνέχειας (δεν ορίζεται η συνάρτηση στο 0).

```
> plot(f(x),x);
```



Σχεδιάζοντας τη γραφική παράσταση της συνάρτησης, παρατηρούμε ότι η εικόνα κοντά στο μηδέν είναι παραπλανητική για τη συνέχεια της συνάρτησης.

Στο παράδειγμα αυτό μπορούμε να άρουμε την ασυνέχεια της f, αν ορίσουμε $f(0) = \lim_{x \to 0} f(x)$.

```
> limit(f(x),x=0);
```

$$1$$

Έτσι, μπορούμε να ορίζουμε τη συνεχή επέκταση της συνάρτησης f, τη συνάρτηση g.

```
> g:=piecewise(x=0,1,f(x));
```

$$g := \begin{cases} 1 & x = 0 \\ \dfrac{\sin(x)}{x} & otherwise \end{cases}$$

□

Άσκηση: 11-4

Να εξετασθεί ως προς τη συνέχεια η συνάρτηση $f(x)=\begin{cases}-2 \text{ αν } x \le -1 \\ 1 \text{ αν } x<2 \\ 2 \text{ αν } x>2\end{cases}$.

Λύση:

Ορίζουμε τη συνάρτηση f.

```
> f:= piecewise( x<=-1, -2, x<2,1, x>2, 2 );
```

$$f := \begin{cases} -2 & x \le -1 \\ 1 & x<2 \\ 2 & 2<x \end{cases}$$

Η γραφική της παράσταση είναι:

```
> plot(f(x),x);
```



Αν εξετάσουμε τη συνάρτηση f ως προς τη συνέχεια χρησιμοποιώντας την εντολή `discont`, δεν έχουμε κάποιο αποτέλεσμα, ωστόσο από τη γραφική παράσταση βλέπουμε τα άλματα ασυνέχεια της f.

```
> discont(f(x),x);
Error, (in discont) cannot determine discontinuities
```

Το γεγονός ότι το πρόγραμμα δεν επιστρέφει σημεία ασυνέχειας, δε σημαίνει ότι η συνάρτηση είναι και συνεχής, μπορούμε να χρησιμοποιήσουμε την εντολή `fdiscont`, η οποία προσπαθεί να βρει σημεία ασυνέχειας αριθμητικά.

Γενική σύνταξη:	***fdiscont(f(x),x)***

Έτσι με την εντολή `fdiscont` στο διάστημα [-3,3] βρίσκουμε:

```
> fdiscont(f(x),x=-3..3);
```

[-1.0002372528699 .. -0.99969064171854, 1.9996883520076 .. 2.0002348816205]

Αν θέλουμε τη γραφική παράσταση της συνάρτησης f, στην οποία να φαίνονται τα σημεία ασυνέχειας, τότε στην εντολή plot χρησιμοποιούμε την παράμετρο `discont=true`.

```
> plot(f(x),x,discont=true);
```



Άσκηση: 11-5

Να εξετασθεί αν η συνάρτηση $f(x)=\left|x-x^2\right|$ παρουσιάζει τοπικό ακρότατο στο διάστημα $\left[-2,2\right]$.

Λύση:

Ορίζουμε τη συνάρτηση.

```
> f:=unapply(abs(x-x^2),x);
```

$$f := x \to \left| -x + x^2 \right|$$

Βρίσκουμε την πρώτη παράγωγο.

```
> diff(f(x),x);
```

$$\mathrm{abs}(1, -x + x^2)\,(-1 + 2\,x)$$

Ορίζουμε τη συνάρτηση της 1ης παραγώγου.

```
> df:= unapply(%,x);
```

$$df := x \to \mathrm{abs}(1, -x + x^2)\,(-1 + 2\,x)$$

Κρίσιμα σημεία θα είναι εκείνα που μηδενίζουν την πρώτη παράγωγο και εκείνα στα οποία δεν ορίζεται η πρώτη παράγωγος.

Λύνουμε την εξίσωση:

```
> solve(df(x)=0,x);
```

$$\frac{1}{2}$$

Αν μετατρέψουμε τη συνάρτηση της πρώτης παραγώγου σε πολύκλαδη συνάρτηση, τότε βλέπουμε ότι αυτή δεν ορίζεται στο x=0 και x=1.

```
> convert(df(x),piecewise);
```

$$\begin{cases} -1+2\,x & x<0 \\ \mathit{undefined} & x=0 \\ 1-2\,x & x<1 \\ \mathit{undefined} & x=1 \\ -1+2\,x & 1<x \end{cases}$$

Άρα, κρίσιμα σημεία είναι τα x= $\frac{1}{2}$, 0,1 .

- Θα εφαρμόσουμε το κριτήριο της δεύτερης παράγωγου, για να προσδιορίσουμε το είδος του πιθανού ακρότατου στο $\frac{1}{2}$.

Η δεύτερη παράγωγος της f είναι:

```
> diff(df(x),x);
```

$$\mathrm{signum}(1, -x+x^2)\,(-1+2\,x)+2\,\mathrm{abs}(1, -x+x^2)$$

Ορίζουμε με d2f την συνάρτηση της δευτέρας παραγώγου

```
> d2f:= unapply(%,x);
```

$$\mathit{d2f} := x \rightarrow \mathrm{signum}(1, -x+x^2)\,(-1+2\,x)+2\,\mathrm{abs}(1, -x+x^2)$$

Η τιμή $f''\left(\frac{1}{2}\right)$ είναι:

```
> d2f(1/2);
```

$$-2$$

Παρατηρούμε ότι $f''\left(\frac{1}{2}\right)<0$. Άρα στο $\frac{1}{2}$ έχουμε τοπικό μέγιστο και η μέγιστη τιμή είναι:

```
> f(1/2);
```

$$\frac{1}{4}$$

- Για να προσδιορίσουμε το είδος των ακρότατων στα σημεία που δεν ορίζεται η παράγωγος, βρίσκουμε το πρόσημο της πρώτης παραγώγου αριστερά και δεξιά από αυτά. Έτσι

```
> df(0.1);
```

$$0.8$$

```
> df(-0.1);
```

$$-1.2$$

Άρα, στο x=0 έχουμε τοπικό ελάχιστο, όμοια και στο x=1.

Επίσης, ακρότατα έχουμε και στα άκρα του διαστήματος, στα οποία έχουμε μέγιστα.

Ας δούμε τώρα τη γραφική παράσταση της f.

```
> plot(f(x),x=-2..2);
```



Το Maple μας παρέχει τη δυνατότητα να προσδιορίσουμε το μέγιστο ή το ελάχιστο μιας έκφρασης κατευθείαν, με τις εντολές `minimize, maximize`.

Περιγραφή:	***Βρίσκει το απόλυτο μέγιστο ή ελάχιστο μιας έκφρασης-συνάρτησης.***	
Γενική σύνταξη:	***minimize(expr, opt1, opt2, ..., optn)*** ***maximize(expr, opt1, opt2, ..., optn)***	
Παράμετροι:	expr	Αλγεβρική έκφραση της οποίας αναζητούμε το μέγιστο ή ελάχιστο.
	x=min..max	Διάστημα για το x.
	y= min..max	Διάστημα για το y.
	location=	Επιστρέφει τη θέση του ακρότατου false ή true (προεπιλογή false).

```
> minimize(f(x),x=-2..2,location=true);
```

$$0, \{[\{x=0\}, 0], [\{x=1\}, 0]\}$$

```
> maximize(f(x),x=-2..2,location=true);
```

$$6, \{[\{x=-2\}, 6]\}$$

Αν θέλουμε να χρησιμοποιήσουμε τις εντολές αυτές για τον προσδιορισμό τοπικών ακρότατων, από τη γραφική παράσταση βρίσκουμε ένα διάστημα στο οποίο βρίσκεται το τοπικό ακρότατο. Εδώ βλέπουμε ότι βρίσκεται στην περιοχή [0,1], οπότε:

```
> maximize(f(x),x=0..1,location=true);
```

$$\frac{1}{4}, \left\{\left[\{x=\frac{1}{2}\}, \frac{1}{4}\right]\right\}$$

□

Άσκηση: 11-6

Να σχεδιάσετε και να υπολογίσετε το εμβαδόν του χωρίου που περικλείεται από τις $f(x)=2-x^2$ και $f(x)=-x$.

Λύση:

Ορίζουμε τη συνάρτηση f και g.

```
> f:=x->2-x^2;
```

$$f := x \to 2-x^2$$

```
> g:=x->-x;
```

$$g := x \to -x$$

Οι συναρτήσεις αυτές τέμνονται στα σημεία που έχουμε τετμημένες αυτές που προκύπτουν από τη λύση του συστήματος $\begin{cases} y=2-x^2 \\ y=-x \end{cases}$.

```
> solve(f(x)=g(x),x);
```

$$-1, 2$$

Ας δούμε τώρα τη γραφική παράσταση των δύο συναρτήσεων.

```
> plot({f(x),g(x)},x=-1.5..2.5,labels=[x,y], title=" f (x)
= 2-x^2, g(x) = -x");
```



Το εμβαδόν που περικλείεται είναι ίσο με $\int_{-1}^{2}\left(f(x)-g(x)\right)dx$. Άρα:

```
> int(f(x)-g(x),x=-1..2);
```

$$\frac{9}{2}$$

□

Άσκηση: 11-7

Να βρεθεί το πολυώνυμο του Taylor δευτέρου βαθμού με κέντρο το $x_0=\pi$ της συνάρτησης $f(x)=x\sin(x)$ και να γίνει μια εκτίμηση του σφάλματος.

Λύση:

Είναι γνωστό ότι το πολυώνυμο το Taylor δίνεται από τη σχέση :

$p_n(x)=f(x_0)+f^{'}(x_0)(x-x_0)+\dfrac{f^{''}(x_0)}{2!}(x-x_0)^2+\ldots+\dfrac{f^{(n)}(x_0)}{n!}(x-x_0)^n$ με υπόλοιπο $R_n(x)=\dfrac{f^{(n+1)}(\xi)}{(n+1)!}(x-x_0)^{n+1}$, με $\xi=\xi(x)\in(x_0,x)$.

Ορίζουμε τη συνάρτηση:

```
> f:=x->x*sin(x);
```

$$f := x \to x \sin(x)$$

Η συνάρτηση της πρώτης παραγώγου είναι:

```
> df:=D(f);
```

$$df := x \to \sin(x) + x\cos(x)$$

Η συνάρτηση της δεύτερης παραγώγου είναι:

```
> ddf:=D(df);
```

$$ddf := x \to 2\cos(x) - x\sin(x)$$

Το πολυώνυμο taylor δευτερης τάξης με κέντρο το π είναι:

```
> pol:=f(Pi)+df(Pi)*(x- Pi)+(ddf(Pi)/2)*(x-Pi)^2;
```

$$pol := -\pi\,(x-\pi) - (x-\pi)^2$$

Το Maple είναι εφοδιασμένο με την εντολή `taylor`, η οποία επιστρέφει το πολυώνυμο.

Θα λύσουμε την παραπάνω άσκηση χρησιμοποιώντας την εντολή `taylor`.

Το ανάπτυγμα του taylor 3ης τάξης με κέντρο το π είναι:

```
> taylor(f(x),x=Pi,3);
```

$$-\pi\,(x-\pi) - (x-\pi)^2 + \mathrm{O}((x-\pi)^3)$$

Ο όρος O που παρουσιάζεται είναι το υπόλοιπο. Μπορούμε να το αφαιρέσουμε χρησιμοποιώντας την εντολή `convert`.

```
> pol:=convert(%,polynom);
```

$$pol := -\pi\,(x-\pi) - (x-\pi)^2$$

Η γραφική παράσταση του πολυωνύμου και της συνάρτησης στο ίδιο σύστημα αξόνων δίνεται ως εξής:

```
> plot([f(x),pol],x=0..2*Pi);
```



Άσκηση:11-8

Δίνεται η καμπύλη $f(x)=\dfrac{x^3}{12}+\dfrac{1}{x}$. Να υπολογισθεί το μήκος της μεταξύ $x=2$ και $x=4$.

Λύση:

Ας θυμίσουμε ότι το μήκος της καμπύλης δίνεται από τη σχέση $L=\int_a^b \sqrt{1+(f'(x))^2}\,dx$

Ορίζουμε τη συνάρτηση.

```
> f:=x->x^3/12+1/x;
```

$$f := x \to \frac{1}{12}x^3+\frac{1}{x}$$

Η γραφική της παράσταση είναι:

```
> plot(f(x),x=0..4,y=0..30);
```



Το μήκος είναι:

```
> int(sqrt(1+(diff(f(x),x))^2),x=2..4);
```

$$\frac{59}{12}$$

□

Άσκηση 11-9

Να βρεθεί ο γεωμετρικός τόπος των σημείων του επιπέδου για τα οποία ισχύει: $\operatorname{Re}\left(\frac{z-i}{z+1}\right)=0$

Λύση:

Ορίζουμε τον αριθμητή και τον παρονομαστή του κλάσματος του οποίου θέλουμε το πραγματικό μέρος.

```
> z1:=x+y*I.-I;
> z2:=x+y*I+1;
```

$$z1 := x + y\,I - I$$
$$z2 := x + y\,I + 1$$

Υπολογίζουμε το κλάσμα.

```
> evalc(z1/z2);
```

$$\frac{x\,(x+1)}{(x+1)^2+y^2}+\frac{(y-1)\,y}{(x+1)^2+y^2}+\left(\frac{(y-1)\,(x+1)}{(x+1)^2+y^2}-\frac{x\,y}{(x+1)^2+y^2}\right)I$$

Θέτουμε ότι τα x, y είναι πραγματικοί αριθμοί.

```
> assume(x,real);assume(y,real);
```

Το πραγματικό μέρος του κλάσματος είναι:

```
> Re(z1/z2);
```

$$\frac{x\sim^2+x\sim+y\sim^2-y\sim}{x\sim^2+2\,x\sim+1+y\sim^2}$$

Ο γεωμετρικός τόπος των σημείων είναι ένας κύκλος τον οποίων θα δημιουργήσουμε με την εντολή `implicit plot`

```
>with( plots ):
implicitplot(Re(z1/z2)=0,x=-3..3,y=-3..3);
```



Κεφάλαιο 12

Λογισμός Συναρτήσεων Πολλών Μεταβλητών

Στο κεφάλαιο αυτό θα δούμε πώς μπορούμε να χρησιμοποιήσουμε το Maple στη μελέτη συναρτήσεων πολλών μεταβλητών. Ουσιαστικά, θα επεκτείνουμε τα όσα αναφέρθηκαν στο κεφάλαιο 10 για τις συναρτήσεις μιας μεταβλητής. Θα ασχοληθούμε με συναρτήσεις πραγματικών τιμών και θα επικεντρώσουμε το ενδιαφέρον μας σε συναρτήσεις δύο και τριών πραγματικών μεταβλητών. Θα ξεκινήσουμε και εδώ δίνοντας τον τρόπο ορισμού αυτών των συναρτήσεων και θα συνεχίσουμε με τη θεμελιώδη έννοια του ορίου. Συνεχίζοντας, θα παρουσιάσουμε τη μερική παράγωγο και την πολλαπλή ολοκλήρωση . Στο τέλος, θα δούμε πώς με τη χρήση του Maple μπορούμε να έχουμε εποπτεία χρησιμοποιώντας τις γραφικές παραστάσεις. Κλείνοντας το κεφάλαιο αυτό, στις λυμένες ασκήσεις θα δούμε αρκετές εφαρμογές των παραπάνω εννοιών, όπως τον υπολογισμό εμβαδού, όγκου κ.ά.

12.1 Ορισμός Συνάρτησης.

Ο ορισμός συνάρτησης πολλών μεταβλητών γίνεται με ανάλογο τρόπο όπως και στις συναρτήσεις μιας μεταβλητής. Ορίζουμε ένα όνομα για τη συνάρτηση (την εξαρτημένη μεταβλητή) και χρησιμοποιούμε το σύμβολο $:=$ ή την `unapply`.

Περιγραφή:	Ορισμός συνάρτησης πολλών μεταβλητών $f\left(x_{1,}x_2,\ldots,x_n\right)$
Γενική σύνταξη:	**Όνομα συνάρτησης:=** **(μεταβλητή1, μεταβλητή2,...)->τύπος**

Παράδειγμα: 12.1

Να ορισθεί η συνάρτηση $f(x,y)=\sin(x)\cos(y)$ και να βρεθεί η τιμή της στο $\left(\pi,\dfrac{\pi}{2}\right)$.

Λύση:

```
> f:=(x,y)->sin(x)*cos(y);
```

$$f := (x, y) \to \sin(x)\cos(y)$$

Καλώντας τη συνάρτηση με το όνομα f θα έχουμε:

```
> f(Pi,Pi/2);
```

$$0$$

□

Παράδειγμα: 12.2

Να ορισθεί η συνάρτηση $f(x,y,z)=x^2+y^2+z^2$ και να βρεθεί η τιμή της στο $(1,0,2)$.

Λύση:

```
> f:=(x,y,z)->x^2+y^2+z^2;
```

$$f := (x, y, z) \to x^2+y^2+z^2$$

```
> f(1,0,2);
```

$$5$$

Στα παραδείγματα που ακολουθούν χρησιμοποιούμε την εντολή `unapply`.

Γενική σύνταξη:	***unapply** (τύπος, μεταβλητή1, μεταβλητή2,...)*

Παράδειγμα: 12.3

Να ορισθεί η $f(x,y)=x^2+y^2$ και να βρεθεί η τιμή της στο (2,2).

Λύση:

```
> f:= unapply(x^2+y^2,x,y);
```

$$f:=(x,y)\rightarrow x^2+y^2$$

```
> f(2,2);
```

$$8$$

Για τον ορισμό πολύκλαδης συνάρτησης μπορούμε και εδώ να χρησιμοποιήσουμε την εντολή `piecewise`.

Περιγραφή:	*Ορίζει μια Πολύκλαδη συνάρτηση.*
Γενική σύνταξη:	*Όνομα συνάρτησης:=piecewise(συνθήκη 1, τύπος 1, συνθήκη2, τύπος2)*

Παράδειγμα: 12.4

Να ορισθεί η $f(x,y)=\begin{cases}1-x^2-y^2 & \text{αν } x^2+y^2<1\\ 0 & \text{αν } x^2+y^2\geq 1\end{cases}$, να βρεθεί η τιμή της το (2,2) και να γίνει η γραφική της παράσταση.

Λύση:

Ορίζουμε τη δίκλαδη συνάρτηση f.

```
> f:=(x,y)->piecewise(x^2+y^2<1,1-x^2-y^2, x^2+y^2>=1,0);
```

$$f:=(x,y)\rightarrow \text{piecewise}(x^2+y^2<1,\, 1-x^2-y^2,\, 1\leq x^2+y^2,\, 0)$$

Η τιμή της στο σημείο (2,2) είναι:

```
> f(2,2);
```

$$0$$

Για τη γραφική της παράσταση θα χρησιμοποιήσουμε την εντολή `plot3d`, την οποία και θα αναλύσουμε με λεπτομέρειες στο επόμενο κεφάλαιο.

```
> plot3d(f(x,y),x=-2..2,y=-2..2);
```



12.2 Όριο

Για τον υπολογισμό των ορίων στις συναρτήσεις πολλών μεταβλητών θα χρησιμοποιήσουμε την εντολή `limit`, όπως την είδαμε στην περίπτωση της μιας μεταβλητής, καθώς και γνωστά θεωρητικά αποτελέσματα για όρια βασικών συναρτήσεων και πράξεων μεταξύ ορίων.

Περιγραφή:	**_Υπολογίζει το Όριο μιας έκφρασης._**	
Γενική σύνταξη:	**_limit(f(x,y),{ x=a, y=b})_**	
Παράμετροι:	f:	Η συνάρτηση της οποίας αναζητούμε το όριο.
	(a,b):	Το σημείο που αναζητούμε το όριο.

Παράδειγμα: 12..5

Να υπολογιστεί το όριο $\lim\limits_{(x,y)\to(1,0)}\left(3x^2+5y^2\right)$.

Λύση:

Ορίζουμε τη συνάρτηση f.

```
> f:=(x,y)->3*x^2+5*y^2;
```

$$f := (x, y) \to 3\,x^2 + 5\,y^2$$

```
> limit(f(x,y),{x=1,y=0});
```

$$3$$

□

Παράδειγμα: 12-6

Να εξεταστεί αν υπάρχει το όριο $\lim\limits_{(x,y)\to(0,0)} \dfrac{x+y}{x-y}$.

Λύση:

Ορίζουμε αρχικά τη συνάρτηση $f(x, y) = \dfrac{x+y}{x-y}$.

```
> f:=(x,y)->(x-y)/(x+y);
```

$$f := (x, y) \to \frac{x-y}{x+y}$$

Ζητώντας το όριο $\lim\limits_{(x,y)\to(0,0)} \dfrac{x+y}{x-y}$ το Maple μας απαντά ότι αυτό δεν υπάρχει.

```
> limit(f(x,y),{x=0,y=0});
```

$$undefined$$

Πράγματι, αν ζητήσουμε τα δύο διαδοχικά όρια $\lim\limits_{x\to0}\left(\lim\limits_{y\to0} f(x,y)\right)$, $\lim\limits_{y\to0}\left(\lim\limits_{x\to0} f(x,y)\right)$, θα δούμε ότι αυτά δεν είναι ίσα.

```
> limit(limit(f(x,y),x=0),y=0);
```

$$-1$$

```
> limit(limit(f(x,y),y=0),x=0);
```

$$1$$

Είναι γνωστό ότι το όριο $\lim\limits_{(x,y)\to(x_0,y_0)} f(x,y)$ εάν υπάρχει είναι ανεξάρτητο της κατεύθυνσης $(x, y) \to (x_0, y_0)$. Αν βρούμε μια κατεύθυνση $(x, y) \to (x_0, y_0)$ και το όριο εξαρτάται από αυτή, συμπεραίνουμε ότι το όριο δεν υπάρχει. Έτσι, αν εδώ θεωρήσουμε την κατεύθυνση $(x,mx) \to (0,0)$, έχουμε:

```
> eval(f(x,y),y=m*x);
```

$$\frac{x - m\,x}{x + m\,x}$$

```
> simplify(%);
```

$$-\frac{-1 + m}{1 + m}$$

Βλέπουμε, λοιπόν ότι το όριο εξαρτάται από το m, δηλαδή από την κατεύθυνση. Άρα, το όριο δεν υπάρχει.

□

12.3 Μερική Παράγωγος

Όπως γνωρίζουμε, μερικές παράγωγοι είναι εκείνες που παίρνουμε όταν σε μια συνάρτηση πολλών μεταβλητών κρατάμε σταθερές όλες τις ανεξάρτητες μεταβλητές εκτός από εκείνη ως προς την οποία παραγωγίζουμε. Έχοντας αυτό υπ'όψιν μας και την ήδη γνωστή εντολή `diff` της παραγώγου μιας μεταβλητής, θα εξετάσουμε τις μερικές παραγώγους συναρτήσεων πολλών μεταβλητών.

Παρακάτω θα δούμε πώς συντάσσονται οι εντολές για τον υπολογισμό των μερικών παραγώγων μιας συνάρτησης δύο μεταβλητών. Ο τρόπος αυτός εύκολα γενικεύεται και για συναρτήσεις περισσότερων μεταβλητών.

Περιγραφή:	***Υπολογίζει την παράγωγο μιας έκφρασης.***	
Γενική σύνταξη:	***diff(expr,x)***	
Παράμετροι:	expr	Η προς παραγώγιση έκφραση.
	x	Η μεταβλητή ως προς την οποία γίνεται η παραγώγιση.
Expression Pallette	$\frac{\partial}{\partial x} f$	

Παράδειγμα: 12-7

Να υπολογιστούν οι μερικές παράγωγοι μέχρι και δευτέρας τάξης της συνάρτησης $f(x,y) = x^2 + y^2$.

Λύση

```
> f:=(x,y)->x^2+y^2;
```

$$f := (x, y) \to x^2 + y^2$$

Οι $\frac{\partial f}{\partial x}, \frac{\partial f}{\partial y}$ είναι:

```
> diff(f(x,y),x);
```

$$2\,x$$

```
> diff(f(x,y),y);
```

$$2\,y$$

Οι $\frac{\partial^2 f}{\partial x^2}, \frac{\partial^2 f}{\partial y^2}$ είναι:

```
> diff(f(x,y),x$2);
```

$$2$$

```
> diff(f(x,y),y$2);
```

$$2$$

Η μεικτή παράγωγος δευτέρας τάξης $\frac{\partial^2 f}{\partial x \partial y}$ είναι:

```
> diff(f(x,y),x,y);
```

$$0$$

□

Παράδειγμα: 12-8

Να υπολογισθούν οι μερικές παράγωγοι δευτέρας τάξης της συνάρτησης $f(x, y) = e^{x \ln y}$.

Λύση

Ορίζουμε τη συνάρτηση f.

```
> f:=unapply(exp(x*log(y)),x,y);
```

$$f := (x, y) \to \mathrm{e}^{(x \ln(y))}$$

Εδώ ,αν θέλουμε να έχουμε το μαθηματικό συμβολισμό της εντολής, γράφουμε το πρώτο γράμμα κεφαλαίο. Έτσι, μπορούμε να έχουμε τα αποτελέσματα:

```
> Diff(f(x,y),x)=diff(f(x,y),x);
```

$$\frac{\partial}{\partial x}\left(e^{(x\ln(y))}\right) = \ln(y)\, e^{(x\ln(y))}$$

```
> Diff(f(x,y),y)=diff(f(x,y),y);
```

$$\frac{\partial}{\partial y}\left(e^{(x\ln(y))}\right) = \frac{x\, e^{(x\ln(y))}}{y}$$

```
> Diff(f(x,y),x,y)=diff(f(x,y),x,y);
```

$$\frac{\partial^2}{\partial y\,\partial x}\left(e^{(x\ln(y))}\right) = \frac{e^{(x\ln(y))}}{y} + \frac{\ln(y)\, x\, e^{(x\ln(y))}}{y}$$

```
> Diff(f(x,y),y,x)=diff(f(x,y),y,x);
```

$$\frac{\partial^2}{\partial x\,\partial y}\left(e^{(x\ln(y))}\right) = \frac{e^{(x\ln(y))}}{y} + \frac{\ln(y)\, x\, e^{(x\ln(y))}}{y}$$

```
> Diff(f(x,y),x,x)=diff(f(x,y),x,x);
```

$$\frac{\partial^2}{\partial x^2}\left(e^{(x\ln(y))}\right) = \ln(y)^2\, e^{(x\ln(y))}$$

```
> Diff(f(x,y),y,y)=diff(f(x,y),y,y);
```

$$\frac{\partial^2}{\partial y^2}\left(e^{(x\ln(y))}\right) = -\frac{x\, e^{(x\ln(y))}}{y^2} + \frac{x^2\, e^{(x\ln(y))}}{y^2}$$

□

Παράδειγμα: 12-9

Να υπολογιστούν οι μερικές παράγωγοι της συνάρτησης $f(x,y)=\dfrac{x^2-y^2}{x^2+y^2}$ και να οριστούν ως συνάρτηση.

Λύση

Αν χρησιμοποιήσουμε την εντολή D (που είδαμε στο κεφάλαιο 9), έχουμε την παράγωγο μιας συνάρτησης δοσμένη και αυτή ως συνάρτηση.

```
> f:=(x,y)->x*y*(x^2-y^2)/(x^2+y^2);
```

$$f := (x,y) \rightarrow \frac{x\,y\,(x^2-y^2)}{x^2+y^2}$$

Η συνάρτηση της πρώτης παραγώγου ως προς x είναι:

```
>fx:=D[1](f);
```

$$fx := (x,y) \rightarrow \frac{y\,(x^2-y^2)}{x^2+y^2} + \frac{2\,x^2\,y}{x^2+y^2} - \frac{2\,x^2\,y\,(x^2-y^2)}{(x^2+y^2)^2}$$

Η συνάρτηση της πρώτης παραγώγου ως προς y είναι:

```
>fy:=D[2](f);
```

$$fy := (x,y) \rightarrow \frac{x\,(x^2-y^2)}{x^2+y^2} - \frac{2\,x\,y^2}{x^2+y^2} - \frac{2\,x\,y^2\,(x^2-y^2)}{(x^2+y^2)^2}$$

Η συνάρτηση της μεικτής παραγώγου ως προς x,y είναι;

```
> fxy:=D[1,2](f);
```

$$fxy := (x,y) \rightarrow \frac{x^2-y^2}{x^2+y^2} + \frac{2\,x^2}{x^2+y^2} - \frac{2\,x^2\,(x^2-y^2)}{(x^2+y^2)^2} - \frac{2\,y^2}{x^2+y^2} - \frac{2\,y^2\,(x^2-y^2)}{(x^2+y^2)^2} + \frac{8\,x^2\,y^2\,(x^2-y^2)}{(x^2+y^2)^3}$$

Έτσι, μπορούμε να υπολογίσουμε την τιμή μιας μερικής παραγώγου σε ένα σημείο, για παράδειγμα στο (1,2).

```
> fx(1,2);
```

$$\frac{2}{25}$$

□

12.4 Κατευθυνόμενη Παράγωγος

Με την εντολή `DirectionalDiff` μπορούμε να υπολογίσουμε την κατευθυνόμενη παράγωγο κατά μια κατεύθυνση σε ένα σημείο, η οποία υπενθυμίζουμε ότι ορίζεται $\left(D_u f\right)_{P_0} = \left(\nabla f\right)_{P_0} \cdot u$ όπου u η κατεύθυνση και P_0 το σημείο.

Περιγραφή:	**Υπολογίζει την κατευθυνόμενη παράγωγο.**	
Γενική σύνταξη:	***DirectionalDiff(f, v, c)***	
Παράμετροι:	f	Η συνάρτηση.
	v	Η κατεύθυνση.
	c	Οι συντεταγμένες.
Πακέτο:	**VectorCalculus**	

Παράδειγμα: 12-10

Να βρεθεί η κατευθυνόμενη παράγωγος της συνάρτησης $f(x, y) = 1 - x^2 - y^2$ στην κατεύθυνση (u,v) και στο σημείο (1,1).

Λύση:

```
> with(VectorCalculus):
> f:=(x,y)->1-x^2-y^2;
```

$$f := (x, y) \rightarrow VectorCalculus{:-}`+`(VectorCalculus{:-}`+`(1, VectorCalculus{:-}`-`(x^2)), VectorCalculus{:-}`-`(y^2))$$

```
> DirectionalDiff(f(x,y),<u,v>, [x,y]);
```

$$\frac{-2\,x\,u - 2\,y\,v}{\sqrt{u^2 + v^2}}$$

Υπολογίζουμε την κατευθυνόμενη παράγωγο στο σημείο (1,1).

```
> eval(%,{x=1,y=1});
```

$$\frac{-2\,u - 2\,v}{\sqrt{u^2 + v^2}}$$

Το ίδιο αποτέλεσμα μπορούμε να έχουμε με την εντολή `DirectionalDerivative` του πακέτου εντολών

`Student[MultivariateCalculus]`.

```
> with(Student[MultivariateCalculus]):
> DirectionalDerivative(f(x,y),[x,y]=[1,1],[u,v]);
```

$$\frac{-2\,u-2\,v}{\sqrt{u^2+v^2}}$$

Αν προσθέσουμε και την παράμετρο output=plot, μπορούμε να έχουμε το ακόλουθο σχήμα όπου φαίνεται κλίση της εφαπτομένης στο σημείο (1,1) κατά την κατεύθυνση (0,1).

```
> DirectionalDerivative(f(x,y),[x,y]=[1,1], [0,1], output=plot);
```



12.5 Πολλαπλή Ολοκλήρωση

Για τον υπολογισμό των πολλαπλών ολοκληρωμάτων χρησιμοποιούμε την εντολή int, που είδαμε στις συναρτήσεις μιας μεταβλητής. Έτσι, ένα διπλό ολοκλήρωμα μπορούμε να το υπολογίσουμε εφαρμόζοντας διαδοχικά δυο φορές την εντολή `int`.

Περιγραφή:	$\int_a^b[\int_c^d f(x,y)dy]dx$	
Γενική Σύνταξη:	***int(int(f(x,y), y=c..d), x=a..b)***	
Παράμετροι:	$f(x,y)$	Η προς ολοκλήρωση συνάρτηση.
	x	a,b άκρα ολοκλήρωσης για το x.
	y	c,d άκρα ολοκλήρωσης για το y.

Παράδειγμα: 12-11

Να υπολογιστεί το διπλό ολοκλήρωμα $\int_0^1 \int_{-1}^2 xe^y dxdy$.

Λύση:

```
> int(int(x*exp(y),x=-1..2),y=0..1);
```

$$-\frac{3}{2}+\frac{3}{2}\mathrm{e}$$

Και εδώ μπορούμε να κάνουμε χρήση του κεφαλαίου γράμματος της εντολής, ώστε να έχουμε το αποτέλεσμα:

```
> Int(Int(x*exp(y),x=-1..2),y=0..1)= int(int( x*exp(y),x=-
1..2),y=0..1);
```

$$\int_0^1 \int_{-1}^2 x\,\mathrm{e}^y\,dx\,dy = -\frac{3}{2}+\frac{3}{2}\mathrm{e}$$

□

Παράδειγμα: 12-12

Με το ίδιο σκεπτικό, δηλαδή χρησιμοποιώντας την εντολή `int`, μπορούμε να υπολογίσουμε και τριπλά ολοκληρώματα.

Στο κεφάλαιο των ασκήσεων θα δούμε διάφορες περιπτώσεις για τον υπολογισμό πολλαπλών ολοκληρωμάτων και θα υπενθυμίσουμε πώς μπορούμε να υπολογίσουμε τα άκρα ολοκλήρωσης σε διάφορες περιπτώσεις.

Παράδειγμα: 12-13

Να υπολογιστεί το τριπλό ολοκλήρωμα $\int_0^1 \int_1^2 \int_{-1}^2 xyzdxdydz$.

Λύση:

```
> Int(Int(Int(x*y*z,x=-1..2),y=1..2),z=0..1)=
int(int(int(x*y*z,x=-1..2),y=1..2),z=0..1);
```

$$\int_0^1 \int_1^2 \int_{-1}^2 x\,y\,z\,dx\,dy\,dz = \frac{9}{8}$$

□

Κεφάλαιο 13

Γραφικές Παραστάσεις στο Χώρο

Στο κεφάλαιο αυτό θα παρουσιάσουμε τον τρόπο με τον οποίων μπορούμε να δημιουργήσουμε γραφικές παραστάσεις στων χώρο. Επίσης θα παρουσιάσουμε τα βασικά εντολές για την δημιουργία των βασικών σχημάτων του ΄χωρου όπως είναι η σφαίρα, ο κύβος, ο κώνος, ο κύλινδρος και ο τόρος

13.1 Γραφική Παράσταση Συνάρτησης Δυο Μεταβλητών

Η εντολή `plot3D` δίνει τη γραφική παράσταση μιας συνάρτησης δύο μεταβλητών στο χώρο. Οι παράμετροι, όπως φαίνονται και στον παρακάτω πίνακα, είναι ανάλογες με αυτές της εντολή `plot`.

<table>
<tr><td>Περιγραφή:</td><td colspan="2">Σχεδιάζει τη γραφική παράσταση μιας συνάρτησης δύο μεταβλητών στο χώρο.</td></tr>
<tr><td rowspan="4">Γενική Σύνταξη:</td><td colspan="2">plot3d(f(x,y), x=xmin..xmax, y=ymin..ymax, παράμετροι)</td></tr>
<tr><td>$f(x,y)$</td><td>Η συνάρτηση δύο μεταβλητών.</td></tr>
<tr><td>xmin..xmax</td><td>Το διάστημα στον άξονα των x.</td></tr>
<tr><td>ymin..ymax</td><td>Το διάστημα στον άξονα των y.</td></tr>
</table>

Παράμετρος	**Επιλογές**
axes=	Ορίζει το είδος των αξόνων. *Δυνατές τιμές:* BOXED, NORMAL, FRAME, and NONE. *Προεπιλογή:* NONE.
labels=[x,y,z]	Ορίζει τις ετικέτες στους άξονες.
style=	Ορίζει πώς θα είναι η επιφάνεια που θα σχεδιαστεί. *Δυνατές τιμές:* POINT, HIDDEN, PATCH, WIREFRAME, CONTOUR, PATCHNOGRID, PATCHCONTOUR, or LINE. *Προεπιλογή:* PATCH
coords=	Ορίζει το είδος των συντεταγμένων. *Προεπιλογή:* καρτεσιανές.
title=	Ορίζει το τίτλο της γραφικής παράστασης. Αν θέλουμε ο τίτλος να καταλαμβάνει περισσότερες από μία γραμμές, στην αρχή κάθε γραμμής πρέπει να εισάγουμε τους χαρακτήρες \n.
tickmarks=[l,n,m]	Ορίζει την αρίθμηση στους άξονες (οι τιμές l,n,m πρέπει να είναι ακέραιοι αριθμοί).

Πίνακας 18. *Παραμέτρων της plot3d*

Παράδειγμα: 13-1

Να γίνει η γραφική παράσταση της συνάρτησης $f(x,y)=\sin(x)\cos(4y)$ για $x\in[-2,2]$, $y\in[-2,2]$.

Λύση:

```
> f:=(x,y)->sin(x)*cos(4*y);
```

$$f:=(x,y)\to \sin(x)\cos(4\,y)$$

Θα χρησιμοποιήσουμε είδος αξόνων `boxed` και θα δώσουμε τίτλο και ετικέτες στους άξονες. Έτσι:

```
>plot3d(f(x,y),x=0..4,y=0..4,axes=boxed,title="sin(x)cos(
4y)",labels=[πλάτος,βάθος,ύψος]);
```



Αν τώρα αλλάξουμε την παράμετρο `style`, τότε η επιφάνεια θα είναι:

```
> plot3d(f(x,y),x=-2..2,y=-2..2,axes=boxed,
title="sin(x)cos(4y)",labels=[πλάτος,βάθος,ύψος],style=PA
TCHNOGRID);
```



Αν τώρα αλλάξουμε την παράμετρο `style=line` και θέσουμε την αρίθμηση στους άξονες ανα δυο , τότε η επιφάνεια θα είναι:

```
> plot3d(f(x,y),x=-2..2,y=-2..2,axes=frame, title=
"sin(x)cos(4y)",labels=[πλάτος,βάθος,ύψος],style=line,
ticmarks=[2,2,2]);
```



□

Παράμετρος	**Επιλογές**
ambientlight=[r,g,b]	Ορίζει το περιδεών χρώμα .
numpoints=n	Ορίζει τον ελάχιστο συνολικό αριθμό των σημείων που θα χρησιμοποιηθούν από την εντολή plot3d.
orientation=[theta,phi]	Ορίζει τις γωνίες θέασης.

Πίνακας 19. *Παραμέτρων της plot3d*

Παράδειγμα: 13-2

Να γίνει η γραφική παράσταση της συνάρτησης $f(x,y)=100-x^2-y^2$ για $x\in[-2,2]$, $y\in[-2,2]$.

Λύση:

Για να δημιουργήσουμε την επιφάνεια $f(x,y)=100-x^2-y^2$, θα χρησιμοποιήσουμε τους συνήθεις άξονες (`axes=NORMAL`). Η απόχρωση της επιφάνειας την ορίζουμε σε αναλογία κόκκινου πράσινου και μπλε με 0.5,0.5,1. Επιπλέον οι άξονες θα είναι σε κλίμακα 10,10,10 και οι γωνίες από την οποία βλέπουμε την επιφάνεια ορίζεται ως 20,100 σε μοίρες. Έτσι, έχουμε:

```
> plot3d(100-x^2-y^2,x=-2..2,y=-2..2,axes=NORMAL,
labels=[x,y,z],ambientlight=[0.5,0.5,1],tickmarks=[10,10,
10],orientation=[20,100]);
```



Εδώ πρέπει να αναφερθεί ότι παρέχεται η δυνατότητα να αλλάζουμε τη γωνία πιο άμεσα. Κρατώντας το αριστερό πλήκτρο του ποντικιού πάνω στη γραφική παράσταση και μετακινώντας το ποντίκι αλλάζει η γωνία.

□

Παράμετρος	Επιλογές
font=[family, style, size]	Ορίζει το είδος της γραμματοσειράς για το κείμενο. *Δυνατές τιμές για την παράμετρο* family *:* TIMES, COURIER, HELVETICA, SYMBOL.
labeldirections=[x,y,z]	Ορίζει την κατεύθυνση της ετικέτας. *Προεπιλογή:* HORIZONTAL.
labelfont=	Ορίζει την γραμματοσειρά της ετικέτας.
titlefont=	Ορίζει την γραμματοσειρά του τίτλου.
axesfont=	Ορίζει το είδος της γραμματοσειράς στους άξονες.

Πίνακας 20. *Παράμετροι της plot3d (Γραμματοσειρές)*

13.2 Γραφική Παράσταση Συνάρτησης Πεπλεγμένης Μορφής

Όταν μια συνάρτηση μας δίνεται σε πεπλεγμένη μορφή, τότε δεν μπορούμε να χρησιμοποιήσουμε την εντολή `plot3D` για τη γραφική της παράσταση (άμεσα τουλάχιστον). Το Maple μέσω του πακέτου εντολών `plots` μας δίνει τη δυνατότητα να σχηματίσουμε μια τέτοια γραφική παράσταση με την εντολή `implicitplot3d`.

Περιγραφή:	***Δημιουργεί την Τρισδιάστατη γραφική παράσταση πεπλεγμενης συνάρτησης***	
Γενική σύνταξη:	**implicitplot3d(*εξίσωση, x=a..b, y=c..d, z=p..q, options*)**	
Παράμετροι:	x=a..b,y=c..d, z=p..q	Τα διαστήματα για τις μεταβλητές x,y,z.
	Options	Ό,τι και στην εντολή plot
Πακέτο:	***plots***	

Η εντολή `implicitplot3d` δέχεται ως παραμέτρους τις περισσότερες από τις παραμέτρους της εντολής `plot3d`.

Παράδειγμα: 13-3

Να παρασταθεί γραφικά η επιφάνεια $x^2 + y^2 + z^2 = 1$ (σφαίρα με κέντρο την αρχή των αξόνων και ακτίνα 1).

Λύση:

```
> with(plots):
implicitplot3d( x^2 + y^2 + z^2 =1,x=-2..2,y=-2..2,z=-
2..2);
```



□

Παράδειγμα: 13-4

Να παρασταθεί γραφικά η επιφάνεια $x^3+y^3+z^3+1=(x+y+z)^3$.

Λύση:

```
> with(plots):
implicitplot3d( x^3 + y^3 + z^3 + 1 = (x + y + z +
1)^3,x=-2..2,y=-2..2,z=-2..2);
```



□

Παράδειγμα: 13-5

Να παρασταθεί γραφικά η επιφάνεια $(x-1)^2+(y+1)^2+z^2=4$ (σφαίρα με κέντρο το σημείο (1,-1,0) και ακτίνα 2).

Λύση:

```
> implicitplot3d( (x-1)^2 + (y+1)^2 + z^2 = 4,x=-1..3,y=-
3..1,z=-2..2,color=red,title="ΣΦΑΙΡΑ", axes=frame);
```



□

13.3 Γραφική Παράσταση Παραμετρικών Εξισώσεων

Για να σχεδιάσουμε την γραφική παράσταση μια συνάρτησης που δίνεται σε παραμετρική μορφή χρησιμοποιούμε την εντολή `plot3d`.

```
> plot3d( [ r*cos(t), r*sin(t), cos(t)*sin(t) ],
r = 0 .. 1, t = 0 .. 2*Pi, grid = [10,60],
orientation=[25,25], shading=zhue, lightmodel=light4);
```



□

13.4 Ισοσταθμικές Καμπύλες

Περιγραφή:	***Σχεδιάζει τις ισοσταθμικές καμπύλες μιας επιφάνειας.***	
Γενική σύνταξη:	**contourplot3d(*f(x,y), x=xmin..xmax, y=ymin..ymax, παράμετροι*)**	
Παράμετροι:	$f(x,y)$	Η συνάρτηση δύο μεταβλητών.
	xmin..xmax	Το διάστημα στον άξονα των x.
	ymin..ymax	Το διάστημα στον άξονα των y.
Πακέτο:	***plots***	

Οι παράμετροι που δέχεται η εντολή `contourplot` είναι ίδιες με εκείνες που δέχεται η εντολή `plot3d`.

Παράδειγμα: 13-6

Να σχεδιαστούν οι ισοσταθμικές καμπύλες της συνάρτησης $f(x,y)=\sin(xy)$.

Λύση:

Ας δούμε αρχικά την επιφάνεια.

```
> plot3d(sin(x*y),x=-3..3,y=-3..3);
```



Ενεργοποιούμε το πακέτο `plots` και κατασκευάζουμε τις ισοσταθμικές καμπύλες.

```
> with(plots):
>contourplot(sin(x*y),x=-3..3,y=-3..3);
```



```
> contourplot(sin(x*y),x=-3..3,y=-3..3, filled=true);
```



□

13.5 Εμφανίζοντας Πολλές Γραφικές Παραστάσεις

Περιγραφή:	***Εμφανίζει στο ίδιο Σύστημα Αξόνων Τρισδιάστατα Σχήματα***
Γενική σύνταξη:	**display3d(*γραφική παράσταση*)** **display3d(*γραφική παράσταση, options*)**
Παράμετροι:	Οι παράμετροι είναι ίδιες με αυτές του plot3d

Παράδειγμα: 13-7

Να σχεδιαστούν στο ίδιο σύστημα αξόνων οι $f(x, y) = x^2 + y^2$, x=0, y=0.

Λύση:

Ενεργοποιούμε το πακέτο εντολών Plots
```
> with(plots):
```
Ορίζουμε την συνάρτηση f
```
>f:=x^2 + y^2:
```
Ονομάζουμε ως p1 την γραφική παράσταση της συνάρτησης
```
>p1:=plot3d(f,x=-2...2, y=-2..2,color=red);
```

Για να δημιουργήσουμε τα επίπεδο x=0 και y=0 χρησιμοποιούμε την εντολή `implicitplot`. Το πρώτο επίπεδο το καλούμε ως p2 και το δεύτερο ως p3.

```
>p2:=implicitplot3d(x=0,x=-3..3,y=-3..3,z=-2..3,color
=green):
>p3:=implicitplot3d(y=0,x=-3..3,y=-3..3,z=-2..3,color
=blue):
```

Με την εντολή `display` εμφανίζουμε τα 3 σχήματα μαζί.

```
>display3d({p1,p2,p3},labels=[x,y,z],axes=NORMAL);
```



13.5 Βασικά Σχήματα στο Χώρο

Στην παράγραφο αυτή θα παρουσιάσουμε τις εντολές εκείνες που μας δίνουν τη δυνατότητα να κατασκευάσουμε τα βασικότερα από τα σχήματα στο χώρο. Για τη χρήση αυτών των εντολών είναι απαραίτητο να ενεργοποιηθούν τα πακέτα `plots` και `plottools`.

ΣΧΗΜΑ	ΕΝΤΟΛΗ
ΣΦΑΙΡΑ	**sphere(*c, r, options*)**
ΚΥΒΟΣ	**cuboid(*a, b, options*)**
ΚΥΛΙΝΔΡΟΣ	**cylinder(*c, r, h, options*)**
ΚΩΝΟΣ	**cone(*c, r, h, options*)**
ΤΟΡΟΣ	**torus(*[x, y, z], r, R, options*)**

Πίνακας 21. *Βασικά Σχήματα στο χώρο*

Παράδειγμα: 13-8

Να σχεδιαστούν:

a. Μια σφαίρα με κέντρο το (0,0) και ακτίνα 3.

b. Ένας κύβος με γωνίες τα σημεία (0,0,0) και (1,1,1).

c. Ένας κύλινδρος με κέντρο της βάσης το (1,1,1) ακτίνα 3 και ύψος 3.

d. Ένας κώνος με κέντρο της βάσης το (0,2,-2) ακτίνα 2 και ύψος 5.

e. Ένας τόρος με κέντρο του τόρου το (1,1,1) ακτίνα 1 και απόσταση 2.

Λύση:

Ενεργοποιούμε τα πακέτα εντολών `plots` και `plottools.`

```
> with(plottools):with(plots):
```

Ορίζουμε τη σφαίρα με κέντρο το (0,0) και ακτίνα 3 και χρησιμοποιώντας την εντολή `display` ζητάμε να εμφανισθεί στην οθόνη.

```
> a:=sphere([1,1,1], 3):
display(a,scaling=constrained, style=patch, axes=boxed);
```



Ορίζουμε τον κύβο και χρησιμοποιώντας την εντολή `display` ζητάμε να εμφανισθεί στην οθόνη.

```
> b:=cuboid([0,0,0],[1,1,1]):
display(b);
```



Ορίζουμε τον κύλινδρο και χρησιμοποιώντας την εντολή `display` ζητάμε να εμφανισθεί στην οθόνη.

```
> c:=cylinder([1,1,1], 3,3):
display(c);
```



Ορίζουμε τον κώνο και χρησιμοποιώντας τη πάλη εντολή `display` ζητάμε να εμφανισθεί στην οθόνη.

```
> d:=cone([0,2,-2],2,5):
display(d);
```



Ορίζουμε τον τόρο και χρησιμοποιώντας τη εντολή `display` ζητάμε να εμφανισθεί στην οθόνη.

```
> e:=torus([1,1,1], 1, 2):
>display(e, scaling=constrained,axes=boxed);
```



Περιγραφή:	***Σχεδιάζει ένα «σωλήνα» γύρο από μια παραμετρική καμπύλη.***
Γενική σύνταξη:	**tubeplot (*έκφραση*)**
Παράμετροι:	Οι παράμετροι είναι ίδιες με αυτές του plot3d
Πακέτο:	**plots**

Θα παρουσιάσουμε ορισμένα παραδείγματα της εντολής tubeplot.
Αρχικά ενεργοποιούμε το πακέτο εντολών plots.

```
> with(plots):
> f:=[(x-5*Pi)*sin(x)/3, (x-5*Pi)*cos(x)/3, (x-5*Pi)*.9,
x=0 .. 5*Pi]:

>tubeplot(f, radius = (x-5*Pi)*.2,tubepoints =30,
style=PATCH,lightmodel=light4, shading=zhue, orientation
= [-90,90]);
```



```
> with(plots):
>tubeplot({[cos(t),sin(t),0],[0,sin(t)-1,cos(t)]},
t=0..2*Pi,radius=1/4,style=patch,lightmodel=light4,
shading=zhue);
```



```
>tubeplot([3*sin(t),t,3*cos(t)],t=-3*Pi..4*Pi,
radius=1.2+sin(t),numpoints=80,lightmodel=light4,
shading=zhue);
```



```
>tubeplot({[cos(t),sin(t),0,t=Pi..2*Pi,numpoints=15,radiu
s=0.25*(t-Pi)],[0,cos(t)-1,sin(t),t=0..2*Pi,
numpoints=45,radius=0.25]},lightmodel=light4,
shading=zhue);
```



□

Κεφάλαιο 14

Λυμένες Ασκήσεις στις Συναρτήσεις Πολλών Μεταβλητών

Στο κεφάλαιο αυτό θα χρησιμοποιήσουμε τις εντολές που είδαμε μέχρι τώρα για να λύσουμε κάποιες χαρακτηριστικές ασκήσεις στις συναρτήσεις δύο μεταβλητών .

Άσκηση: 14-1

Να ορισθεί η συνάρτηση $f(x,y)=\sqrt{y-x^2}$, και να παρασταθεί γραφικά το πεδίο ορισμού της καθώς και η επιφάνεια που αυτή ορίζει στο χώρο.

Λύση:

Ορίζουμε τη συνάρτηση με όνομα μεταβλητής f .

```
> f:=(x,y)->sqrt(y-x^2);
```

$$f:=(x,y)\to\sqrt{y-x^2}$$

Η συνάρτηση ορίζεται όταν $y - x^2 \geq 0$.

```
> solve(y-x^2>=0,y);
```

$$\{x^2 \leq y\}$$

Ένας τρόπος για να παραστήσουμε το κομμάτι του επιπέδου που ορίζεται από την παραπάνω ανίσωση είναι:

```
> with(plots):
> implicitplot(y=x^2,x=-1..1,y=0..1,filled=true);
```



Η γραφική παράσταση της f είναι:

```
> plot3d(f(x,y),x=-2..2,y=-2..2,axes=normal);
```



□

Άσκηση: 14-2

Να βρεθούν, αν υπάρχουν, τα ακρότατα της συνάρτησης $f(x,y)=x^3-3x^2+y^2$.

Λύση:

Ορίζουμε τη συνάρτηση f.

```
> f:=(x,y)->x^3-3*x^2+y^2;
```

$$f:=(x,y)\to x^3-3\,x^2+y^2$$

Υπολογίζουμε τις παραγώγους πρώτης τάξης $\frac{\partial f}{\partial x}(x,y)$ και $\frac{\partial f}{\partial y}(x,y)$, τις οποίες τις συμβολίζουμε με dfx, dfy αντίστοιχα.

```
> dfx:=diff(f(x,y),x);
```

$$dfx:=3\,x^2-6\,x$$

```
> dfy:=diff(f(x,y),y);
```

$$dfy:=2\,y$$

Βρίσκουμε τα κρίσιμα σημεία λύνοντας το σύστημα $\begin{cases}\frac{\partial f}{\partial x}(x,y)=0\\ \frac{\partial f}{\partial y}(x,y)=0\end{cases}$

```
> ks:=solve({dfx=0,dfy=0},{x,y});
```

$$ks:=\{y=0,x=0\},\ \{y=0,x=2\}$$

Το παραπάνω σύστημα έχει δύο λύσεις. Απομονώνουμε τις λύσεις αυτές και τις συμβολίζουμε με (x1,y1) και (x2,y2).

```
> x1:=rhs(ks[1][2]);
```

$$x1:=0$$

```
> y1:=rhs(ks[1][1]);
```

$$y1:=0$$

```
> x2:=rhs(ks[2][2]);
```

$$x2:=2$$

```
> y2:=rhs(ks[2][1]);
```

$$y2 := 0$$

Υπολογίζουμε τις παραγώγους δευτέρας τάξης.

```
> dfxx:=diff(f(x,y),x$2);
```

$$dfxx := 6\,x - 6$$

```
> dfyy:=diff(f(x,y),y$2);
```

$$dfyy := 2$$

```
> dfxy:=diff(f(x,y),x,y);
```

$$dfxy := 0$$

Σχηματίζουμε την ορίζουσα $\begin{vmatrix} \dfrac{\partial^2 f}{\partial x^2}(x_0, y_0) & \dfrac{\partial^2 f}{\partial x \partial y}(x_0, y_0) \\ \dfrac{\partial^2 f}{\partial x \partial y}(x_0, y_0) & \dfrac{\partial^2 f}{\partial y^2}(x_0, y_0) \end{vmatrix}$ την οποία και συμβολίζουμε με τη μεταβλητή orizousa.

```
> orizousa:=dfxx*dfyy-dfxy^2;
```

$$orizousa := 12\,x - 12$$

Υπολογίζουμε την ορίζουσα στο πρώτο κρίσιμο σημείο (x1,y1).

```
> eval(orizousa,{x=x1,y=y1});
```

$$-12$$

Άρα, στο κρίσιμο αυτό σημείο θα έχουμε ακρότατο. Υπολογίζουμε την $\dfrac{\partial^2 f}{\partial x^2}(x_0, y_0)$ στο (x1,y1).

```
> eval(dfxx,{x=x1,y=y1});
```

$$-6$$

Η ορίζουσα είναι αρνητική. Άρα, στο (x1,y1) έχουμε ελάχιστο.

Τώρα υπολογίζουμε την ορίζουσα στο δεύτερο κρίσιμο σημείο (x2,y2)

```
> eval(orizousa,{x=x2,y=y2});
```

$$12$$

Η ορίζουσα εδώ είναι θετική. Άρα, στο σημείο αυτό έχουμε σαγματικό σημείο.

Ας δούμε τώρα τη γραφική παράσταση της f.

```
> plot3d(f(x,y),x=-1..4,y=-3..3, axes=normal,
orientation=[60,70]);
```



Άσκηση: 14-3

Να υπολογίσετε το επίπεδο που εφάπτεται στο γράφημα της συνάρτησης $f(x,y)=x^2+y^2$ στο σημείο $(0,1)$ και να παραστήσετε αυτό γραφικά.

Λύση:

Ορίζουμε καταρχήν τη συνάρτηση.

```
> f:=(x,y)->x^2+y^2;
```

$$f:=(x,y)\to x^2+y^2$$

Βρίσκουμε τις μερικές παραγώγους ως προς x και y.

```
> dfx:=(x,y)->diff(f(x,y),x);
```

$$dfx:=(x,y)\to \mathrm{diff}(\mathrm{f}(x,y),x)$$

```
> dfy:=(x,y)->diff(f(x,y),y);
```

$$dfy:=(x,y)\to \mathrm{diff}(\mathrm{f}(x,y),y)$$

Γνωρίζουμε ότι το εφαπτομενικό επίπεδο μιας συνάρτησης $f:R^2\to\mathbb{R}$ στο σημείο (x_0,y_0) ορίζεται από την εξίσωση

$z = f(x_0, y_0) + f_x(x_0, y_0)(x - x_0) + f_y(x_0, y_0)(y - y_0)$. Έτσι ορίζουμε με tanplane το εφαπτομενικό επίπεδο όπως αυτό δίνεται από τον τύπο της θεωρίας.

```
> tanplane:=f(x0,y0)+dfx(x0,y0)*(x-x0)+ dfy(x0,y0)*(y-y0);
```

$$tanplane := x0^2 + y0^2 + 2\,x0\,(x - x0) + 2\,y0\,(y - y0)$$

Αντικαθιστούμε το σημείο $(x_0, y_0) = (0,1)$

```
> tanplane:=eval(tanplane,{x0=0,y0=1});
```

$$tanplane := -1 + 2\,y$$

Σχεδιάζουμε το επίπεδο και τη συνάρτηση στο ίδιο σύστημα αξόνων.

```
> plot3d({tanplane,f(x,y)},x=-3..3,y=-3..3,axes=boxed);
```



Άσκηση: 14-4

Να υπολογιστεί το διπλό ολοκλήρωμα $\iint_D xydA$, όπου D είναι το τρίγωνο που ορίζεται από τις ευθείες $y = x + 4$, $y = -x + 2$ και $y = -\frac{1}{3}x - 4$.

Λύση

Ορίζουμε τις εξισώσεις που ορίζουν την περιοχή ολοκλήρωσης D και τις συμβολίζουμε με eq1, eq2, eq3.

```
> eq1:=y=x+4;
```

$$eq1 := y = x + 4$$

```
> eq2:=y=-x+2;
```

$$eq2 := y = -x + 2$$

```
> eq3:=y=-(1/3)*x-4;
```

$$eq3 := y = -\frac{x}{3} - 4$$

Βρίσκουμε τα κοινά σημεία και τα ονομάζουμε sol1, sol2, sol3.

```
> sol1:=solve({eq1,eq2},{x,y});
```

$$sol1 := \{x = -1, y = 3\}$$

```
> sol2:=solve({eq1,eq3},{x,y});
```

$$sol2 := \{x = -6, y = -2\}$$

```
> sol3:=solve({eq2,eq3},{x,y});
```

$$sol3 := \{x = 9, y = -7\}$$

Είναι απαραίτητο να απομονώσουμε μόνο τα δεξιά μέλη των παραπάνω λύσεων. Έτσι, ορίζουμε τα σημεία τομής των ευθειών με (x1,y1), (x2,y2), (x3,y3).

```
> x1:=rhs(sol1[1]);
```

$$x1 := -1$$

```
> y1:=rhs(sol1[2]);
```

$$y1 := 3$$

```
> x2:=rhs(sol2[1]);
```

$$x2 := -6$$

```
> y2:=rhs(sol2[2]);
```

$$y2 := -2$$

```
> x3:=rhs(sol3[1]);
```

$$x3 := 9$$

```
> y3:=rhs(sol3[2]);
```

$$y3 := -7$$

Στις μεταβλητές txt1,txt2,txt3 σημειώνουμε τα σημεία αυτά, για να τα εμφανίσουμε στη συνέχεια στο σχήμα.

```
> with(plots):
```

```
> txt1:=textplot([x1,y1,{x1,y1}]);
```

$$txt1 := \mathrm{PLOT}(\mathrm{TEXT}([-1., 3.], "\{-1, 3\}"))$$

```
> txt2:=textplot([x2,y2,{x2,y2}]);
```

$$txt2 := \mathrm{PLOT}(\mathrm{TEXT}([-6., -2.], "\{-6, -2\}"))$$

```
> txt3:=textplot([x3,y3,{x3,y3}]);
```

$$txt3 := \mathrm{PLOT}(\mathrm{TEXT}([9., -7.], "\{-7, 9\}"))$$

Με grd συμβολίζουμε τη γραφική αναπαράσταση των τριών εξισώσεων.

```
> grd:=implicitplot({eq1,eq2,eq3,x=-1},x=-8..10,y=-9..5):
```

Τώρα, θα σχεδιάσουμε την περιοχή ολοκλήρωσης D, η οποία είναι το τρίγωνο που σχηματίζεται από τις 3 ευθείες.

```
> display(grd,txt1,txt2,txt3);
```



Έτσι, η περιοχή ολοκλήρωσης D μπορεί να χωρισθεί σε δύο «x-απλά» τρίγωνα.

$$D_1 = \left\{ (x,y) \in \mathbb{R}^2 : -6 \le x \le -1, -\frac{1}{3}x - 4 \le y \le x + 4 \right\}$$

$$D_2 = \left\{ (x,y) \in \mathbb{R}^2 : -1 \le x \le 9, -\frac{1}{3}x - 4 \le y \le -x + 2 \right\}$$

Και το ζητούμενο ολοκλήρωμα είναι:

$$\iint_D xydA = \iint_{D_2} xydA + \iint_{D_2} xydA = \int_{-6}^{-1}\int_{-\frac{1}{3}x-4}^{x+4} xydydx + \int_{-1}^{9}\int_{-\frac{1}{3}x-4}^{-x+2} xydydx$$

Άρα χρησιμοποιώντας την εντολή int έχουμε:

```
> int(int(x*y,y=-(1/3)*x-4..x+4),x=-6..-1)+
int(int(x*y,y=-(1/3)*x-4..-x+2),x=-1..9);
```

$$-275$$